
            Rebol в примерах

1. Создание GUI

Наберите код ниже в редакторе и сохраните с расширением .r:

view layout/size [] 400x300

Код программы создаст окно 400 пикселей поперек, и 300 пикселей вниз (пиксели -
точки на компьютерном экране). Окно GUI может быть перемещено вокруг экрана,
свернуто и закрыто ("X" в правом верхнем углу), точно так же как любая другая
программа Windows. В других языках программирования, только создание окна подобно
этому может занять несколько страниц кода.

Чтобы добавлять кнопку к вышеупомянутому GUI, наберите следующий код. Обратите
внимание, что слово "button" добавилось между скобками:

view layout/size [button] 400x300

Теперь графический интерфейс пользователя имеет универсальную синюю кнопку, по
которой вы можете щелкнуть мышью. Чтобы добавить некоторый текст к кнопке,
напечатайте следующий код. Обратите внимание, что текст "Нажми" добавился после
button:

view layout/size [button "Нажми"] 400x300

Чтобы после нажатия на кнопку произошло действие, наберите следующий код, и затем
щелкните кнопкой. Обратите внимание, что текст ["Привет Мир"] добавился после текста
кнопки:

view layout/size [button "Нажми" [alert "Привет Мир!"]] 400x300

Теперь, когда вы щелкаете кнопкой в вашем окошке, компьютер отвечает, выдавая
сообщение "Привет Мир!". Давайте добавим больше интерактивности, наберите следующий
код, и затем щелкните кнопкой еще раз. Обратите внимание на "data: request-text"
добавление в начале строки. Этот код запрашивает некоторый текст от пользователя и
назначает это слову "data":

data: request-text view layout/size [
button "Нажми" [alert data]] 400x300

Код выше разбит на две строки так, чтобы это вместилось на страницу учебника, но это
может быть набрано в интерпретатор Ребола как одна строка. Одна строка - все, что
требуется, чтобы создать программу, которая получает некоторые данные от
пользователя, создает GUI, который ждет пользовательского взаимодействия, и
обрабатывает входные данные (отображает это в небольшом диалоговом окне).

Затем, мы сохраним некоторые данные на жесткий диск вашего компьютера. Наберите
следующий код, и щелкните "yes", если интерпретатор Ребола просит у вас разрешения

http://www.rebol.com/


сохранить на жесткий диск. Обратите внимание на запись “%/c/data.txt data"
добавленный в конец строки. Код запишет данные, полученные от пользователя, в файл
на диск C: , названный "data.txt”

data: request-text view layout/size [
button "Нажми" [alert data]] 400x300 write %/c/data.txt data

Если вы наблюдаете за C: вашего компьютера после закрытия окна, вы увидите, что там
теперь существует текстовый файл (C:\data.txt) содержащий текст, который вы
напечатали в своей программе. (Если Вы работаете в другой операционной системе кроме
Windows, вы должны изменить "С:" символ в вышеупомянутой строке, чтобы обратиться к
корневому каталогу на вашем жестком диске).

С одной только строкой, вы имеете рабочую программу, которая делает что-то полезное.
Получает, отображает, и сохраняет некоторые данные от пользователя, используя
знакомые взаимодействия GUI. Можно модифицировать код, чтобы хранить телефонные
номера, имена пользователя и пароли, или любую другую полезную информацию.

Преимущество Rebol в том, он может иметь дело со всеми типами общих данных. Вы
можете легко отображать фотографии и другую графику в ваших GUI, web-страницах, и
т.д. Вот код, который загружает изображение с сервера и отображает в GUI:

view layout [image load http://rebol.com/view/bay.jpg]

Слово "image" в скобках отображает изображение в GUI. Слово "load" загружает
изображение, которое будет отображено.

Rebol позволяет многим встроенным эффектам быть примененным к изображениям. Обратите
внимание на код "effect [effect type]"  в следующих примерах:

view layout [image load http://rebol.com/view/bay.jpg effect [Grayscale]]
view layout [image load http://rebol.com/view/bay.jpg effect [Emboss]]
view layout [image load http://rebol.com/view/bay.jpg effect [Flip 1x1]]

Вы можете получить данные с сервера и сохранять на вашем диске C:. "/binary"
модификатор используется всякий раз, когда имеет дело с двоичными данными:

write/binary %/c/bay.jpg read/binary http://rebol.com/view/bay.jpg

Теперь вы можете читать изображение непосредственно с вашего жесткого диска и
отображать в GUI. Наберите код ниже. Обратите внимание на "view layout", и [image
load %/c/bay.jpg], файл загружен с диска C:

view layout [image load %/c/bay.jpg]

2. Больше примеров

Ниже - некоторые более короткие примеры чтения, записи, и управления данными
относительно жесткого диска и Internet, и взаимодействия с пользователем. Наберите
их в интерпретатор Rebol.

Следующая строка отображает текущую дату и время:



print now

Слово "print" отображает текстовые данные непосредственно в интерпретаторе Rebol.
Слово "now" обращается к текущей дате и времени.

Следующая строка исполняет некоторые математические вычисления, и отображает
результат:

print (10 + 12) / 2

Следующий код просит, чтобы пользователь выбрал файл на жестком диске:

request-file

Код ниже позволяет пользователю выбирать цвет:

request-color

Следующий код спрашивает пользователя:

request "Вы согласны?"

Позволить пользователю выбирать дату:

request-date

Код ниже запрашивает имя и пароль пользователя:

request-pass

Следующий код открывает web-браузер вашего компьютера и отображает указанную web-
страницу:

browse http://rebol.com

Следующий код запускает встроенный текстовый редактор, и открывает файл c:\test.txt

editor %/c/test.txt

Обратите внимание на символ процента ("%") в примере выше. В Rebol, этот символ
используется, чтобы представить все метки файла. Поскольку Rebol может использоваться
на многих операционных системах, чтобы обратиться к дискам, путям, и т.д., Rebol
использует универсальный формат: %/drive/path/path/.../file.ext. Например,
"%/c/Windows/notepad.exe" обращается к "C:\Windows\Notepad.exe" в Windows. Rebol
конвертирует тот синтаксис к соответствующему формату операционной системы, так,
чтобы ваш код мог быть написан однажды и использоваться на каждой операционной



системе, без последующих изменений.

Следующая строка посылает электронную почту user@website.com, содержа текст "Привет
пользователь. Чем занимаетесь? ". Пробуйте заменить имя пользователя и сайт с вашим
собственным адресом электронной почты (Если вы загрузили и выполнили Rebol,  вы
будете должны выполнить конфигурацию, чтобы послать электронную почту.):

send user@website.com "Привет пользователь. Чем занимаетесь?"

Строка ниже посылает web-страницу пользователю:

send user@website.com read http://www.rebol.com

Код ниже отображает содержание почтового ящика пользователя:

print read pop://user:pass@website.com

Следующая строка загружает единственный файл на сервер, использующий ftp:

write/binary ftp://user:pass@website.com read/binary %file

Следующее загружает полный каталог файлов на сервер:

foreach file load %./ [if not dir? file [write/binary join
    ftp://user:pass@website.com/ file read/binary file]]

3. Быстрое сравнение

Чтобы узнать насколько проще идея Rebol, чем другие языки, рассмотрим короткий
пример. Самый простой код, чтобы создать основное окно GUI, был представлен ранее:

view layout/size [] 400x300

Код для того же самого простого примера представлен ниже, на популярном языке
программирования "C++".:

#include <windows.h>

/*  Declare Windows procedure  */
LRESULT CALLBACK WindowProcedure (HWND, UINT, WPARAM, LPARAM);

/*  Make the class name into a global variable  */
char szClassName[ ] = "C_Example";

int WINAPI
WinMain (HINSTANCE hThisInstance,
         HINSTANCE hPrevInstance,
         LPSTR lpszArgument,
         int nFunsterStil)

{



    HWND hwnd;
    /* This is the handle for our window */
    MSG messages;
    /* Here messages to the application are saved */
    WNDCLASSEX wincl;
    /* Data structure for the windowclass */

    /* The Window structure */
    wincl.hInstance = hThisInstance;
    wincl.lpszClassName = szClassName;
    wincl.lpfnWndProc = WindowProcedure;
    /* This function is called by windows */
    wincl.style = CS_DBLCLKS;
    /* Catch double-clicks */
    wincl.cbSize = sizeof (WNDCLASSEX);

    /* Use default icon and mouse-pointer */
    wincl.hIcon = LoadIcon (NULL, IDI_APPLICATION);
    wincl.hIconSm = LoadIcon (NULL, IDI_APPLICATION);
    wincl.hCursor = LoadCursor (NULL, IDC_ARROW);
    wincl.lpszMenuName = NULL;
    /* No menu */
    wincl.cbClsExtra = 0;
    /* No extra bytes after the window class */
    wincl.cbWndExtra = 0;
    /* structure or the window instance */
    /* Use Windows's default color as window background */
    wincl.hbrBackground = (HBRUSH) COLOR_BACKGROUND;

    /* Register window class. If it fails quit the program */
    if (!RegisterClassEx (&wincl))
        return 0;

    /* The class is registered, let's create the program*/
    hwnd = CreateWindowEx (
           0,
            /* Extended possibilites for variation */
           szClassName,
            /* Classname */
           "C_Example",
            /* Title Text */
           WS_OVERLAPPEDWINDOW,
            /* default window */
           CW_USEDEFAULT,
            /* Windows decides the position */
           CW_USEDEFAULT,
            /* where the window ends up on the screen */
           400,
            /* The programs width */
           300,
            /* and height in pixels */
           HWND_DESKTOP,
            /* The window is a child-window to desktop */
           NULL,
            /* No menu */
           hThisInstance,
            /* Program Instance handler */
           NULL
            /* No Window Creation data */
           );

    /* Make the window visible on the screen */
    ShowWindow (hwnd, nFunsterStil);

    /* Run the message loop.
        It will run until GetMessage() returns 0 */
    while (GetMessage (&messages, NULL, 0, 0))
    {
        /* Translate virtual-key messages
            into character messages */



        TranslateMessage(&messages);
        /* Send message to WindowProcedure */
        DispatchMessage(&messages);
    }

    /* The program return-value is 0 -
        The value that PostQuitMessage() gave */
    return messages.wParam;
}
/*  This function is called by the Windows
        function DispatchMessage()  */

LRESULT CALLBACK
WindowProcedure (HWND hwnd, UINT message,
    WPARAM wParam, LPARAM lParam)
{
    switch (message)
    /* handle the messages */
    {
        case WM_DESTROY:
            PostQuitMessage (0);
                /* send a WM_QUIT to the message queue */
            break;
        default:
            /* for messages that we don't deal with */
            return DefWindowProc (hwnd, message,
                wParam, lParam);
    }

    return 0;
}

Вернемся лучвше к Rebol...

4. Понимание переменных и Функций

Все языки программирования используют две важных особенности: переменные и функции.
Функции - команды, которые говорят, что компьютеру делать. Они подобны глаголам в
разговорном языке. Вы использовали некоторые функции уже в более ранних примерах.
Например, функция "print", представляет некоторое действие.

Переменные - имена, присвоенные данным. Они подобны существительным в разговорном
языке. Например, "now" можно рассматривать как переменную. Это - существительное,
обращающееся к части данных.

Функции и переменные должны использоваться в синтаксисе, определенном в соответствии
с языком программирования, который вы используете. Вы должны записать команды более
определенным способом - используя синтаксис, который неизменно содержит переменные и
функции. Например, чтобы вывести картинку с помощью своей программы, необходимо:

1. назначить имя файла картинки для переменной (существительное).

2. использовать функцию (глагол), чтобы отобразить изображение, упомянутое
переменной.

3. назначить функциональное действие на кнопку или некоторый другой графический
элемент в GUI.

Rebol высокоуровневый язык, поскольку близок к человеческому, но все еще требует
определенного синтаксиса и структуры. Вы должны узнать, как использовать переменные
и функции в грамматике, которую язык определяет. Это основа программирования.

5. Слова

Если вы хотите дать метку некоторым данным - так, чтобы это могло использоваться в
вашей программе, вы должны назначить им слово. Есть много служебных сло, которые



представляют общие действия ("print", "alert", "request-date", и т. Д.). Вы видели
множество встроенных слов уже в более ранних примерах. Чтобы создавать ваши
собственные функции, вы берете предварительно созданные служебные слова и
переменные, группируете их вместе в определенном порядке, и назначаете слово на эту
совокупность кода. Тогда вы можете обратиться к этой группе действий или связанных
данных, используя слово, которое вы назначили для своей функции.

Изучение служебных слов, которые уже определены в языке программирования -
необходимость для изучения языка. Слово "write", например, записывает данные на
запоминающее устройство (жесткий диск, CD диск, сервер, и т.д.). Это - функция,
которая выполняет действие, и ожидает имя для файла, который будет записан на
жесткий диск:

write %/c/text.txt "Здесь набранный вами текст"

Если вы набрали вышеупомянутую строку в неправильном порядке, это не будет работать:

write "Здесь набранный вами текст" %/c/text.txt ; WRONG

Если вы напечатали это так, Rebol не поймет синтаксис, и вы получите ошибку.

6. GUI, Слова и Грамматика – еще глубже

Вы видели в более ранних примерах, что слова "view layout", сопровождаемые двумя
скобками (" [] ") могут использоваться, чтобы отобразить GUI в Rebol. Вы можете
поместить элементы, которые вы хотите видеть в GUI, в скобки. Rebol содержит слова,
которые отображают все обычные графические элементы, используемые в GUI. Пробуйте
набрать код ниже:

view layout [button]

view layout [field]

view layout [text "Rebol действительно очень прост для программирования"]

view layout [text-list]

view layout [
button
field
text "Rebol действительно очень прост для программирования"
text-list

    check
]

Обратите внимание, что слова могут быть отделены "незаполненным пространством" в
скобках. Дополнительные пространства, переводы каретки, и другие пустые символы
игнорируются интерпретатором. Табуляторы традиционно используются, чтобы выровнять
строки в пределах скобок, но они не требуются.

Более наглядные характеристики о графических элементах могут быть включены
непосредственно после каждого из их соответствующих слов. Такие модификаторы
называют "facets", и они позволяют вам корректировать все характеристики каждого
типа графического элемента (размер, цвет, отображение текста), и т.д. Пробуйте
набрать код ниже - это то же самое как вышеупомянутый код, с некоторыми
дополнительными изменениями:

view layout [
button red "Нажми"



field "Введите здесь текст"
text "Rebol действительно очень прост для программирования" purple
text-list 400x300 "Строка 1" "Строка 2" "другая строка"
check yellow

]

6.1 Действия

ВАЖНО: Если вы хотите, чтобы графический элемент исполнил действие,  помещайте текст
в скобки после указания действия. Когда элементом GUI щелкают мышью или иначе,
действие будет выполнено. Наберите следующий код, чтобы увидеть, как это работает:

view layout [button [alert "Вы щелкнули кнопкой."] ]
view layout [button red "Нажми" [alert "Вы щелкнули кнопкой."]]
view layout [

text "Некоторые действия.  Попробуйте использовать виджет:"
button red "Нажми" [alert "Вы щелкнули кнопкой."]
field 400 "Наберите здесь текст, затем нажмите Enter."

        [alert value]
text-list 400x300 "Выбрать строку" "Эта строка" "Новая строка"

        [alert value]
check yellow [alert "Вы щелкнули желтым переключателем."]
button "Выйти" [quit]

]

Обратите внимание, как слово "value" обращается к выбранному элементу в текстовом
списке, и к тексту, содержавшемуся в текстовом поле.

Дополнительные действия (то есть, вызванные правым щелчком мыши) могут быть включены
для любого графического элемента. Только включите их во второй блок, окруженный
скобками:

view layout [button [alert "левый клик"] [alert "правый клик"]]

Вот некоторые другие элементы GUI, используемые в языке Rebol:

view layout [
    backcolor white

h1 "Примеры GUI:"
box red 500x2
bar: progress

    slider 200x16 [bar/data: value show bar]
area "Введите текст"

    drop-down
across
toggle "Нажми" "Здесь" [print value]
rotary "Нажми" "Еще" "И еще" [print value]
choice "Выбрать" "Пункт 1" "Пункт 2" "Пункт 3" [print value]
radio radio radio
led

    arrow
return
text "Normal"
text "Bold" bold
text "Italic" italic
text "Underline" underline
text "Bold italic underline" bold italic underline
text "Стиль текста Serif" font-name font-serif
text "Раздельный текст" font [space: 5x0]
return

    h1 "Заголовок 1"



h2 "Заголовок 2"
    h3 "Заголовок 3"

h4 "Заголовок 4"
tt "Текст пишущей машинки"
code "Текст кода"
below
text "Большой" font-size 32

    title "Центрированный заголовок" 200
    across

vtext "Normal"
vtext "Bold" bold
vtext "Italic" italic
vtext "Underline" underline
vtext "Bold italic underline" bold italic underline
vtext "Стиль текста Serif" font-name font-serif
vtext "Раздельный текст" font [space: 5x0]
return
vh1 "Видео заголовок 1"
vh2 "Видео заголовок 2"
vh3 "Видео заголовок 3"
vh4 "Видео заголовок 3"
label "Метка"
below
vtext "Большой" font-size 32

    banner "Баннер" 200
]

Примеры выше демонстрируют, как Rebol создает GUI, которые могут использоваться,
чтобы ввести и отображать данные. Это основная часть строительства типичной
современной компьютерной программы! Для подробной информации о дизайне GUI, см.
http://rebol.com/docs/easy-vid.html и http://rebol.com/docs/view-guide.html.

7. Создание собственных переменных слов

Точно так же как разговорные языки, языки программирования легки и выразительны. Нет
только одного способа создать программу. Вы должны выбрать и использовать ваши
собственные слова, и вы должны организовать их. Точно так же как на разговорном
языке, вы должны думать на этом языке. В Rebol вы имеете дополнительную способность
создать ваши собственные слова языка, чтобы выразить действия и маркировать данные.

Слова создаются и назначаются для переменных и функций в Rebol при помощи двоеточия
(":") . Вы можете использовать любое слово, которое вы хотите определить к любым
данным или действиям. Например, если вы хотите использовать слово "picture", чтобы
обратиться к загрузочному модулю в Internet, вы можете сделать следующее:

picture: load http://rebol.com/view/bay.jpg

Вышеупомянутая строка создает метку, которая может использоваться в вашей программе,
где вы хотите читать и использовать файл, расположенный в
http://rebol.com/view/bay.jpg. Встроенное Rebol слово "load" найдет изображение по
указанному пути.

Теперь вы можете использовать слово "picture", чтобы обратиться к вышеупомянутому
изображению. Отобразите это в GUI, используя следующий код:

view layout [image picture]

Слова "view", "layout", и "image" встраивают в Rebol, и слово "image" теперь также
допустимо, потому что это было определено для интерпретатора.

Поскольку слово "picture" является переменным, вы можете также переопределить и



изменить данные, упомянутые здесь:

picture: load http://rebol.com/view/demos/palms.jpg

Теперь, когда вы используете слово "picture" в вашей программе, оно обращается к
разным файлам в разных местах Internet. Запись того же самого кода GUI теперь
отображает разные изображения:

view layout [image picture]

Вы можете также сделать так чтобы слово "picture", обращалось к файлам на вашем
жестком диске, или где-нибудь еще, где вы хотели бы:

picture: load %/c/bay.jpg

Вот еще некоторые примеры создания и использования переменных слов. Напечатайте их в
Rebol интерпретатор, чтобы видеть, как они работают, и понимать, как назначенные
слова могут обращаться к любым данным. ОБРАТИТЕ ВНИМАНИЕ: после точки с запятой, код
игнорируется интерпретатором. Это используется, чтобы включить комментарии в код:

acolor: "синий"

alert acolor ; в диалоговом окне текст синий

print acolor ; выводит слово "синий" в Rebol интерпретаторе

anumber: 12

print anumber ; выводит число 12 в Rebol интерпретаторе

computation: (10 + 12) / 2

print computation ; выводит ответ

filename: request-file

print filename ; выводит имя пользователя

chosen-color: request-color

print chosen-color

answer: request "Здесь текст?"

print answer

pick-a-date: request-date

print pick-a-date

userpass: request-pass

print userpass

webpage: http://rebol.com

browse webpage

file: %./test.txt

editor file



; открывает встроенный в Rebol текстовый редактор и в нем же файл
; указанный выше

email-address: user@webpage.com

message:  "Привет друг. Как дела?"

send email-address message

8. Блоки

В Rebol, вы можете использовать слова, чтобы представить множественные части данных
вместе, совокупности других переменных слова, и других элементов программирования.
Заключите данные в скобках. Наберите код ниже, чтобы видеть, как это работает:

somecolors: ["красный" "желтый" "синий" "черный"]
; "somecolors" теперь определенное слово, используется, чтобы представить весь
; блок данных, включенный в скобках.
print somecolors

В Rebol, совокупности подобно этому называют "блоками", и они очень важны.
Фактически, они - первичный организационный модуль в Rebol и главной структуре, в
которой сохранены данные.  В Rebol, для любого типа данных может быть назначено
слово, и блоки могут содержать любую комбинацию слов и необработанных данных. Вы уже
использовали блоки, чтобы отображать элементы GUI и исполнять действия в них. Любая
группа слов, окруженных скобками формирует блок, и вы можете назначить слово к этому
блоку. Слова, относящиеся к сложным блокам данных могут аналогично группироваться в
другие блоки. Это делает работу с очень сложными структурами данных, очень простой в
Rebol.

Наберите код ниже, чтобы видеть, как полное GUI может быть сформировано и
представлено, используя одну метку слова:

gui-layout: [button field text-list]

"gui-layout" теперь обращается к  полному блоку. Вы можете отобразить это используя
"view layout":

view layout gui-layout

Блоки данных могут содержать более чем несколько строк, и могут быть отделены
дополнительным незаполненным пространством. Только заключите элементы в скобки:

gui-layout2: [
    button red "Нажми"

field "Введите текст."
text " Rebol действительно очень прост для программирования." purple
text-list 400x300 "строка 1" "строка 2" "другая строка"
check yellow

]

Теперь отобразите это в GUI:

view layout gui-layout2



Блоки, содержащие в себе другие блоки, содержат данных намного больше, чем в одном
слове!

ОБРАТИТЕ ВНИМАНИЕ: Это - стандартная практика, чтобы выровнять составные блоки с
последовательными табуляторами. Начальные и конечные скобки типично помещаются на
том же самом стандартном уровне. Это обычно в большинстве языков программирования,
потому что это делает сложный код более легким для чтения. Например, составной блок
ниже:

[blue red green [1 2 4 [jan feb march [monday tuesday wednesday]]]]

может быть написан немного более ясно как:

[blue red green
    [1 2 4
        [jan feb march
            [monday tuesday wednesday]
        ]
    ]
]

Сдвиг не требуется, но очень полезно когда имеете дело с большим количеством
запутанных структур.

Вот - простой пример таблицы данных, содержит информацию в пределах блока:

schedule: [
    ["Андрей" "Понедельник" "3:00 pm"]
    ["Саша" "Вторник" "11:00 am"]
    ["Маша" "Среда" "4:45 pm"]
]

Вы можете отобразить, вышеупомянутый блок в GUI используя встроенное Rebol слово
"list". Обратите внимание на код ниже - вставлен полный блок списка показанный выше.

    vh2 "Дежурные этой недели:"
    list 600x400 [
        across text 150 text 150 text 100
    ] [ data schedule
]

Графическое приложения "базы данных", в несколько строк программы... Нет проще!

Встроенные слова помогают вам управлять данными, сохраненными в блоках. Наберите
следующий код, чтобы видеть, как работает слово "sort":

somecolors: ["красный" "желтый" "синий" "черный"]

sortedcolors: sort somecolors

"Sort" сформирована в служебном слове, которое в алфавитном порядке сортирует
элементы данного блока. Строка выше создает недавно определенное слово
"sortedcolors", и назначает это на сортированный блок слов, содержавшихся в
"somecolors".



print sortedcolors
; Этот код отображает сортированный блок текста.

print first sortedcolors
; "first" - другое встроенное слово.
; выбирает первый элемент в данном блоке.
find somecolors "красный"
; "find" является сформированным в слове, которое ищет данные в пределах блока.

Вы можете легко сохранить блоки данных на ваш жесткий диск, читать их, и выполнять
другие операции.

write %/c/colors.txt somecolors
; пишет полный блок текста, представленного "somecolors"
; в текстовый файл по имени colors.txt на C: диск.

Интересный прием, демонстрирует, как Rebol может легко смешать datatypes в пределах
блока:

an-image: load http://rebol.com/view/bay.jpg
; загружает изображение из Internet и назначает
; ему слово "an-image".

append sortedcolors an-image
; "append", добавляет загруженное изображение до конца блока данных
; содержа простые текстовые слова, определенные выше.
; Новый блок содержит и текст и  двоичные данные изображения,
; весь блок назначен на единственное слово!

Теперь вы можете выбрать элементы из нового блока:

print first sortedcolors
; печатает первый элемент в блоке данных - текст "черный".

view layout [image fifth sortedcolors]
; отображает пятый элемент в блоке данных-
; изображение, загруженное выше - в GUI .

При работе с последовательными данными в блоках (тип данных называют "series" в
Rebol):

view layout [image sortedcolors/5]
; "sortedcolors/5" - другой способ обратиться к пятому элементу
; в блоке данных.

Вышеупомянутое приложение работает другим способом:

length-of-block: length? sortedcolors
; встроенный слово "length? " возвращает число
; в блоке (5 в этом регистре).
view layout compose [image sortedcolors/(length-of-block)]



Слово "compose", позволяет вставлять переменные в круглые скобки.

Следующие примеры демонстрируют, что дополнительные слова могут использовать
последовательный ряд данных в пределах блока:

insert sortedcolors "сиреневый"
; добавляет слово "сиреневый" к блоку данных sortedcolors.

remove sortedcolors
; удаляет первый элемент из блока.

head sortedcolors
; устанавливает маркер позиции в начале блока данных.

next sortedcolors
; устанавливает маркер позиции в следующем элементе в блоке данных.

last sortedcolors
; устанавливает маркер позиции в последнем элементе в блоке данных.

back sortedcolors
; устанавливает маркер позиции в предыдущем элементе в блоке данных.

tail sortedcolors
; устанавливает маркер позиции после последнего элемента в блоке данных

Вы можете смешать вместе все типы данных в пределах блока, обращается к частям
блоков по имени, обращается и изменять данные с помощью встроенных функций. Блоки и
переменные слова, назначенные на блоки помогают вам хранить, управлять, и обращаться
ко всем данным, с которыми вы будете иметь дело в ваших программах.

9. Служебные слова

Создание новых функций сопоставимо созданию ваших собственных слов глагола на
разговорном языке. Только будьте осторожны, чтобы не использовать слова, которые уже
определены в языке Rebol, или в вашей текущей программе. Это изменило бы значение
существующего слова. Например, вы можете случайно изменить, значение слова "write",
чтобы обратиться к изображению на жестком диске, печатая следующее:

write: read %/c/bay.jpg
; *** не используйте это - это изменит значение слова
; "write" в Rebol интерпретаторе (только для текущего сеанса).
; Это пример того, как не надо делать. ***

Вы можете защитить все встроенные Rebol слова, используя "protect-system". Это
выдаст предупреждение с ошибкой и отвергнет любые попытки переопределить служебные
слова Rebol. Вы все еще должны быть осторожны, чтобы случайно не переопределить
слова, которые вы создали. Если это произошло, то надо сделать рестарт
интерпретатора Rebol.

Вот действительно важная концепция: В Rebol, вы можете использовать отдельные слова,
чтобы представить полные блоки действий (то есть, совокупности сгруппированных
служебных слов). Фактически, блоки подобно этой форме составляют программы на языке
Rebol. Вот пример нескольких служебных слов, сгруппированных в блок (заключены в
скобки), и новое служебное слово:

some-actions: [
alert "Первое действие."
print "Второе действие."
write %/c/anotheraction.txt "Третье действие."



]

Вышеупомянутый код создал своего рода суперглагол, который обращается к нескольким
действиям. Вы можете исполнить действия, содержавшиеся в любом блоке, используя
служебное слово "do" . Чтобы получить все действия в вышеупомянутом блоке кода,
наберите:

do some-actions

Вы можете также включить слово "does" в определение слова - будет делать действия,
содержавшиеся в блоке, исполняя автоматически каждый раз, когда новое слово
используется в Rebol:

more-actions: does [
alert "4"
alert "5"
alert "6"

]

Фактически, включением "does", вы только что создавали новую функцию (или
"подпрограмма"), которая может использоваться подобно любой другой встроенной в
Rebol! Вы можете теперь общаться с интерпретатором, используя это слово. После того,
как вы ввели код выше, пробуйте набрать "more-actions" в Rebol интерпретатор:

more-actions

Вот пример действия, чтобы очистить командную строку в Rebol интерпретаторе.

cls: does [prin "^(1B)[J"]

Родной способ очистки командной строки интерпретатора: набрать "prin" ^ (1B) [J".
Это неудобно делать каждый раз. Вместо этого, мы можем поручить слову "cls"
исполнять действие - точно так же как в Basic. Теперь только нанаберите:

cls

и экран чист - это намного проще.

Вот небольшая программа, которая создает новое слово, "send-email". Это создаст
простой текстовый интерфейс для пользователей, чтобы послать электронную почту:

send-email: does [
email-address: to-email request-text/title/default

        "Введите адрес email:" "user@webpage.com"
; вышеупомянутая строка создает новое переменное слово "email-address"

        ; "email-address" назначено значение
        ; текстовый ввод, используя встроенный слово "request-text"
        ; "title" и "default"  настраивают
        ; отображение информации в текстовом поле.

message: request-text
; строка выше создает новое переменное слово "message"

        ; и назначает на некоторый требуемый текст
send email-address message

; строка выше посылает сообщение



        ; на адрес электронной почты, данный ранее
alert "Ваше сообщение послано."

]

send-email ; сделайте подпрограмму выше
send-email ; сделайте это снова, чтобы послать другое сообщение кому - то еще

Вышеупомянутый процесс ОЧЕНЬ важен. Это основа того, как вы будете исполнять более
сложные действия в Rebol. Блоки действий и блоки данных формируют основание
программы. В Rebol вы только группируете биты данных вместе в блоки, назначаете имя
и обращаются к нему. Вы также группируете функции вместе в блоки, чтобы выполнить
действия, используя данные, назначаете имя, чтобы через него обратиться к тем
действиям. Вы можете даже комбинировать законченные группы данных и функций в блоки,
которым могут быть назначены уникальные идентификаторы слова, которые исполняют
законченные программные задачи для вас. Это позволяет вам создавать ваш собственный
уникальный язык в любой программе, которую вы пишите, используя слова, которые вы
определяете. Слова не встраиваются в Rebol, но им могут быть назначены
соответствующее действие и значения данных, чтобы сделать  предложение полностью
функциональной частью кода, который понимает компьютер. Создание своих слов
называют, это ваш диалект в Rebol, и это - одна из особенностей Rebol, отличающего
его от других языков. Другие языки сосредотачиваются на различных способах
группировать и управлять функциями и переменными. Как правило, другие языки
используют более сложные способы. Например, блоки, которые содержат и полностью
скрытые функции, и переменные, называют "объектами".

10. Несколько способов создания функций в Rebol

Есть несколько встроенных слов в языке Rebol , которые позволяют вам создавать более
сложные служебные слова. Чтобы создавать простые функции, вы можете использовать
команду "does" , как описано выше. Но некоторые функции более сложны, чем это. Они
работают с разными данными. Например, следующая простая функция отображает
квадратный корень 4.:

sqr-four:  does [print square-root 4]

Слово "sqr-four" теперь назначено на действие "вычислить квадратный корень 4" (слово
"print square-root 4" встраивают в Rebol). После ввода вышеупомянутой строки в Rebol
интерпретаторе, напечатайте:

sqr-four

Получим  2.

Предположим, что мы хотим сделать кое-что, используя другие числа кроме 4...
Например, нужно создать функцию, которая добавляет 4 к некоторому другому числу, и
затем вычисляет квадратный корень той суммы? Имя переменной надо тогда "передать-
passed" к функции. Встроенное слово "func" используется, чтобы создать функции, к
которым изменяемые переменные можно передать. Синтаксис для слова "func" ожидает,
что это сопровождается двумя блоками кода. Первый блок содержит названия переменных,
которые будут переданы. Второй блок содержит действия, которые будут предприняты.
Вот как это выглядит:

func [имена переменных, которые будут переданы] [
действия, которые будут предприняты к тем переменными

]

В следующей строке, создана функция, в которой вызывают "anumber". Функции присвоено
имя "sqr-var":



sqr-var: func [anumber] [print square-root (4 + anumber)]

Теперь можно использовать слово "sqr-var", при этом Rebol интерпретатор знает, что
сделать с назначенными данными. Пробуйте следующий код:

sqr-var 12 ; печатает "4", квадратный корень 12+4 (16)
sqr-var 96 ; печатает "10", квадратный корень 96+4 (100)

Вычисление квадратного корня 4+somenumber не очень интересно, но это помогло
иллюстрировать один из самых важных методов, используемых во всех современных языках
программирования. Процесс принятия параметров переменных  для функций -
фундаментальная часть всего современного программирования. Это возможно
единственный, самый общий элемент в современных языках, и понимание этого, является
необходимым.

Следующая строка создает простую функцию для вывода изображения:

display: func [filename] [view layout [image load filename]]

Это принимает имя файла изображения как переданный параметр (%somefile.jpg,
%somefile.gif, %somefile.png, или %somefile.bmp), и затем создает графический
интерфейс пользователя, чтобы отобразить изображение. Этот набор действий назначен
на слово "display":

image1: to-file request-file/title trim {
Выберите изображение на  жестком диске:} ""

; получает имя файла изображения от пользователя

display image1
; отображает вышеупомянутое изображение, используя новое служебное слово

display http://rebol.com/view/bay.jpg
; отображает изображение в вышеупомянутом url

display %/c/bay.jpg
; отображает изображение, которое было сохранено на
; жестком диске в примере ранее

Вот пример, который спрашивает у пользователя 2 URL сайта, и затем открывает их в
отдельных окнах браузера:

openwebsite: func [nameofwebsite] [browse nameofwebsite]
; Строка выше создает новую функцию, которая передает url
; к встроенному в Rebol слову "browse", чтобы открыть переданный
; сайт в заданном по умолчанию браузере пользователя. Также
; назначает новое служебное слово "openwebsite" на тот набор
; действий.

website1: request-text/title "Введите URL сайта:"
; Строка выше назначает новое переменное слово "website1" для текста
; возвращенного встроенной функцией "request-text".

website2: request-text/title "Введите URL другого сайта:"
; Получает еще некоторый текст и назначает новое переменное слово "website2"
; к этому.

openwebsite website1



; Строка выше использует новое "openwebsite" определенное служебное слово
; , и пути к "website1".

openwebsite website2
; Использует функцию openwebsite снова, на сей раз передавая
; переменную website2.

В этом примере, слово "openwebsite" назначено на новое функциональное определение.
"Website1" и "website2" - метки для переменных. Ниже пример, в котором назначаем
единственное слово на весь полный процесс:

display-website: does [
openwebsite: func [nameofwebsite] [browse nameofwebsite]
website: request-text/title "Введите URL сайта:"
openwebsite website

]

Теперь вы можете использовать слово "display-website" в ваших программах, чтобы
выполнить весь блок кода.

display-website

Привыкание к вышеупомянутому синтаксису и мышлению является необходимым. Помните,
иметь дело со всеми типами данных - главная работа, которую вы делаете как
программист. Принятие переменных данных к функциям, которые вы создаете - главный
способ, которым вы выполняете действия с данными во всей вашей программе. В Rebol
легко иметь дело с данными, используя встроенную поддержку общих типов данных.
Понимание того, как вводить, управлять, и выводить эти данные - ваша главная цель.
Использование функций и переменных как описано выше - фундаментальная часть
обучения.

11. Условные операторы

Давая компьютеру, разнообразие действий, основанных на разнообразии ожидаемых
условий, является фундаментальной методикой программирования, используемой во всех
языках.

Математические операторы типа = <> <> (равный, менее - чем, "больше чем", не -
равный) часто используются, чтобы выполнить операции сравнения. Наберите следующий
код:

if now/time > 12:00 [alert "После полудня."]
; now/time - измененная, или "усовершенствованная" встроенная
; функция "now", которая возвращает только текущее время.

Вот более сложный пример:

daily-calories: to-integer request-text/title {Сколько калорий вы употребили
          сегодня?}
; получает некоторую информацию от пользователя и назначает переменное слово
; "daily-calories". Встроенная функция "to-integer"
; проверяет, что Rebol интерпретирует ту информацию как число.
; " {} " символы, окружающие текст заголовка работают
; так же как кавычки, но позволяет охватить
; несколько строк. Очень важный.

if daily-calories > 2500 [alert "Сегодня не ешьте."]



Встроенное в Rebol слово "either" выбирает выполнить или нет между двумя блоками
функций, основанных на условной оценке. Его синтаксис:

either {condition} [
выполнить блок, если условие true

][
    выполнить блок, если условие false
]

Вот пример:

either now/time > 8:00am [alert "Пришло время вставать!"] [
alert "Вы может спать дальше."]

Вот разновидность вышеупомянутого примера:

wake-up: to-time request-text/title "Во сколько вы хотите проснуться?"

either now/time > wake-up [alert "Пришло время вставать!"] [
alert {Вы может спать дальше.}]

Встроенное Rebol слово "switch" выбирает выполнить или нет между многочисленными
функциями, основанными на условных оценках. Его синтаксис:

switch/default {main value} [
{value 1} [выполнить блок если значение 1 = main value]
{value 2} [выполнить блок если значение 2 = main value]
{value 3} [выполнить блок если значение 3 = main value]
; и.т.д...
] [выполнить по умолчанию блок кода, если ни одно из значений не соответствует]

Вы можете сравнить много значений с основным значением, и выполнить блок кода для
каждого значения соответственно. Вот - пример:

favorite-day:  request-text/title "Каков ваш любимый день недели?"

switch/default favorite-day [
    "Понедельник" [alert "Понедельник день тяжелый..."]
    "Вторник" [alert "Вторник тоже не очень..."]
    "Среда" [alert "В среду разгар работы!"]
    "Четверг" [alert "Выходной не за горами..."]
    "Пятница" [alert "Можно начинать расслабляться!"]
    "Суббота" [alert "Кураж во весь рост!"]
    "Воскресенье" [alert "Болит голова!"]
] [alert "Вы не выбрали день!"]

Оператор "Switch" используется часто, потому что программы требуют многочисленные
оценки выбора результата.

Rebol включает богатый набор слов и функциональных структур, которые помогают вам
оценивать условия во всех типах ситуаций, и со всеми типами данных. Понимание, как
использовать их – одно из главных направлений в изучении языка.



12. Циклы

Программы часто проверяют условия и выполняют функции неоднократно. Фактически, в
большинстве больших приложений компьютер часто исполняет циклы для повторной работы,
какой либо функции. Например, в программе напоминания, приложение должно непрерывно
проверять время и дату, чтобы видеть, нужно ли пользователю напомнить о чем-то в
настоящее время. В других типах программ, компьютер, возможно, нуждается в просмотре
условий через совокупности данных, или неоднократно запрашивает и отвечает
пользовательскому запросу. Чтобы обрабатывать такие ситуации, "loop" структуры
систематически повторяют действия в программе.

Встроенное слово "forever" создает простой цикл. Его синтаксис:

forever [блок действий для повторения]

Следующий код создает простой таймер, который предупреждает пользователя, когда
проходит одна минута. Использует цикл, чтобы непрерывно проверять время.

alarm-time: now/time + 60
; назначим переменную на время 60 секунд с этого времени
forever [if now/time = alarm-time [alert "прошло 60 секунд" break]]

Заметьте слово "break" в примере выше. Этот оператор используется для остановки
цикла,  как только отобразится сообщение, иначе цикл будет работать бесконечно.

Вот - более диалоговая версия. Сердце программы - цикл:

event-name: request-text/title "О чем вам напомнит?"
; запрашивает напоминание от пользователя
seconds: to-integer request-text/title trim {

Через какое время напомнить?}
; запрос на количество времени ожидания
alert join "сейчас " [

now/time ", будьте готовы в " seconds " seconds."
]
; отобразить сообщение
alarm-time: now/time + seconds
; установите сигнальное время
forever [

if now/time = alarm-time [
alert join " сейчас " [

alarm-time ", и " seconds
            " прошли секунды.  Это время для: " event-name
            ]

break
    ]
]
; цикл непрерывно сравнивает сигнальное время набора с
; текущим временем, затем отображает предупреждение, когда оно соответствует.

Заметьте "join" слово, используемое в нескольких строках выше. Формат синтаксиса:

join { данные} [блок  данных]

Используя этот формат, соединяем вместе переменные, текст, блоки и другие данные,
чтобы они могли быть вместе использованы, отображены, и задействованы, чтобы
формировать единую часть данных. Создается особый блок данных, собирающий все



индивидуальные элементы в одну часть данных. Синтаксис:

rejoin [item1 item2 item3 ...]

Теперь, назад к циклам. Есть простой цикл, который отображает и обновляет текущее
время в GUI:

view layout [
timer: field

    button "ПУСК" [
        forever [

set-face timer now/time
            wait 1
        ]
    ]
]

Вышеупомянутое GUI содержит два элемента: текстовое поле, которому назначена
переменная метка "timer", и кнопка со словом " ПУСК ". Блок действия для кнопки
содержит цикл, выполняет 2 действия в цикле (пока пользователь не закроет GUI):
встроенное слово "set-face" показывает текущее время в поле "timer", программа ждет
1 секунду, и затем повтор.

Подобно большинству языков, Rebol включает разнообразие функций и с программным
управлением структур, которые позволяют вам образовывать циклы через блоки данных и
исполнять операции, использующие последовательно измененные значения. Обычное
название оператора цикла  на многих языках - "for". Это позволяет вам определять
стартовое значение, значение окончания, возрастающее значение, и имя переменной,
чтобы держать текущее значение, так, чтобы вы могли образовать циклы через
последовательное изменение значений управляемым способом. Вот основной синтаксис для
оператора цикла "for":

for {переменное слово, содержащее значение} {запускающий значение}
{заканчивающий значение} {возрастающее значение} [блок кода для
исполнения, который может использовать текущее переменное значение]

Здесь простые примеры. Наберите их, чтобы увидеть, как они работают:

for counter 1 10 1 [print counter]
; starts on 1 and counts to 10 by increments of 1

for counter 10 1 -1 [print counter]
; starts on 10 and counts backwards to 1 by increments of -1

for counter 10 100 10 [print counter]
; starts on 10 and counts to 100 by increments of 10

for counter 1 5 .5 [print counter]
; starts on 1 and counts to 5 by increments of .5

for timer 8:00 9:00 0:05 [print timer]
; starts at 8:00am and counts to 9:00am by increments of 5 seconds

for dimes $0.00 $1.00 $0.10 [print dimes]
; starts at 0 cents and counts to 1 dollar by increments of a dime

for date 1-dec-2005 25-jan-2006 8 [print date]
; starts at December 12, 2005 and counts to January 25, 2006
; and by increments of 8 days

for alphabet #"a" #"z" 1 [print alphabet]



; starts at the character a and counts to z by increments of 1 letter

Обратите внимание, что Rebol может легко перебрать различные типы данных, может
автоматически увеличивать дату, время, и т.д. На других языках, это сделать сложнее.

Обратите внимание на использование слова "prin"  в последнем примере. Он работает
как "print",  автоматически не вставляет символ возврата каретки (то есть, он
печатает каждый  символ вывода последовательно рядом с предыдущим).

Вот примера цикла "for" , который отображает первые 5 имен файла в текущей папке на
вашем жестком диске:

files: read %.
; получает список текущего каталога,
; и назначает тот блок имен файла к переменным "files"

for count 1 5 1 compose [print files/(count)]
; печать начал с 1-ым элементом в блоке,
; и подсчитывает к 5-ому элементу.

В примере выше, "files/1" - синтаксис, представляющий первый элемент в списке файла,
"files/2" представляет второе, и так далее. Обратите внимание на слово "compose" ,
используемое для цикла.

"Foreach" - другая полезная структура выполнения цикла. Он позволяет вам легко
перебирать совокупность данных. Его синтаксис, выглядит следующим образом:

foreach {имя переменной, ссылающееся на каждый последовательный
элемент в данном блоке} [данный блок] [блок функций, которые будут
выполнены на каждый элемент в данном блоке, используя имя переменной,
чтобы сослаться на каждый элемент последовательно]

Пример ниже печатает имя каждого файла в текущем каталоге на вашем жестком диске:

folder: read %.
; получает список текущего каталога,
; и назначает тот блок имен файла к переменной "folder"

foreach file folder [print file]
; перебирает каждое имя файла, содержавшееся в блоке "folder"
; и печать каждый последовательно.

Следующая строка читает и печатает каждое последовательное сообщение в почтовом
блоке пользователя:

foreach mail read pop://user:pass@website.com [print mail]

"While" другая полезная структура выполнения цикла, используемая в большинстве
языков программирования. Он неоднократно выполняет условную оценку, и затем
выполняет блок программы, пока условие истинно. Синтаксис:

while [условие] [
блок функций, которые будут выполнены, пока условие истинно

]



Вот простой пример:

x: 1 ; создайте значение
while [x <= 5] [

alert to-string x
    x: x + 1
]

Читаем код "x первоначально равен 1. Пока x меньше чем или равен 5, отобразить
значение x, и прибавить 1 к значению x" (то есть, программа отображает количество от
1 до 5).

ОТМЕТЬТЕ: В Rebol, код "x: x + 1" прибавляет 1 к текущему значению x. Одно из обычно
используемых выражений во всем программировании, найденном во всех видах ситуаций
выполнения цикла. Обратите внимание также на слово "to-string". Он конвертирует
значение числа "x" к тексту ("string"). Это требуется, потому что слово "alert"  в
Rebol  отображает только типы данных строки.

Вот некоторые дополнительные примеры цикла "while":

while [not request "Закончить программу сейчас?"] [
alert "Нажмите ДА, чтобы закончить программу."

]

В вышеупомянутом примере "not" полностью изменяет значение данных, полученных от
пользователя (то есть, yes становится no).

alert "Пожалуйста выберите дату" while [request-date <> now/date] [
alert join "Пожалуйста выберите дату" [now/date]]

while [request-pass <> ["секретный" "пароль"]] [
alert " Имя пользователя и пароль являются 'секретными'"]

Пример ниже использует несколько циклов, чтобы напомнить пользователю, когда
покормить кошку, каждые 6 часов между 8am и 8pm. Используется цикл, чтобы увеличить
время, для напоминания, цикл с условием продолжения, чтобы непрерывно сравнивать
прошедшее временя с текущим временем, и цикл, чтобы делать то же самое каждый день,
непрерывно.

    for timer 8:00am 8:00pm 6:00 [
        while [now/time <= timer] [wait .5]
        ; ничего не делайте, пока не время
        ; кормить кошку. "Ждут.5" действий дают
        ; пользователю шанс закрыть программу.

[alert join "Сейчас" now/time ".  Пришло время кормить."]
; если цикл с условием продолжения вышел, пришло время

        ; кормить кошку. После приведения в готовность, возвратитесь и ждите
        ; следующего кормления.
    ]
]

13. Работа с более сложными примерами

Большинство программ более длинно, чем примеры, которые показаны в этом учебнике.
Ниже пример более сложной программы. Он позволяет пользователю использовать Web-
камеру, для отображения видео у себя. Он отображает видео изображения в GUI, у
которого есть несколько кнопок, используемых, чтобы управлять размером,



местоположением, и действием на экране. Всякий раз, когда вы сохраняете программу
Rebol в текстовом файле, код должен начинаться со следующего текста:

REBOL []

Этот текст говорит интерпретатору Rebol, что файл содержит программу Rebol. Он
должен быть включен в начале любой сохраненной программы Rebol. Вы должны включить
дополнительную документацию о программе, например, заголовок и информацию о версии в
квадратных скобках, но она не обязательна.

Примечание: этот пример включает многие слова Rebol и методики, которые еще не были
обсуждены в учебнике. Цель примера состоит в том, чтобы просто предоставить более
длинный текстовый пример, который может быть вырезан, вставлен, и сохранен в файл.
Не беспокойтесь, если пока не все понятно.

Rebol [Title: "Средство просмотра Webcam"]

; try http://www.webcam-index.com для некоторых ссылок webcam.

temp-url: "http://type-some-website-here.com"
while [true]  [

webcam-url: to-url request-text/title/default trim {
Введите URL:} temp-url

either attempt [webcam: load webcam-url]
        [break]
        [either request [trim {

Этот webcam  в настоящий момент не доступен.} trim {
Повторить} "Выйти"]

            [temp-url: to-string webcam-url]
            [quit]
    ]
]
resize-screen: func [size] [

webcam/size: to-pair size
window/size: (to-pair size) + 40x72
show window

]
window: layout [

across
btn "Стоп" [webcam/rate: none show webcam]
btn "Старт" [

webcam/rate: 0
webcam/image: load webcam-url
show webcam

    ]
rotary "320x240" "640x480" "160x120" [

resize-screen to-pair value
    ]

btn"Выход" [quit] return
webcam: image load webcam-url  320x240
with [

        rate: 0
feel/engage: func [face action event][

switch action [
time [face/image: load webcam-url show face]

            ]
        ]
    ]
]
view center-face window

Используйте программу, например, XpackerX http://www.marmaladefoo.com, чтобы
упаковать и распространить программу. XpackerX позволяет вам упаковать интерпретатор
Rebol и webcam.r в отдельный исполняемый файл. Это позволяет вам создавать отдельные



Windows-программы в одном файле.

ВАЖНО: Чтобы выключить используемое по умолчанию значение, которое непрерывно
спрашивает право доступа к чтению и записи на жесткий диск, напечатайте "secure
none" в интерпретаторе Rebol, и затем выполните программу с "do {filename}r".
Выполнение "C:\rebol.exe -s {filename}" делает то же самое. "-S" запускает
интерпретатор Rebol без любых включенных средств защиты, заставляя его вести себя
как типичная Windows-программа.

Использование XpackerX особенно полезно (не только для Rebol, но и для других
интерпретируемых языков). Он избавят ваших пользователей от необходимости загружать
интерпретатор Rebol. Упаковывая интерпретатор Rebol,  сценарий, и файлы поддержки
данных  в отдельную выполнимую программу, работает как компилятор Rebol, который
создает обычные Windows-программы. Чтобы это сделать, вы должны создать файл
"template.xml":

<?xml version="1.0"?>
<xpackerdefinition>
    <general>
        <!--shown in taskbar -->
        <appname>your_program_name</appname>
        <exepath>your_program_name.exe</exepath>
        <showextractioninfo>false</showextractioninfo>
        <!-- <iconpath>c:\icon.ico</iconpath> -->
    </general>
    <files>
        <file>
            <source>your_rebol_script.r</source>
            <destination>script.r</destination>
        </file>
        <file>
            <source>C:\Program Files\rebol\view\Rebol.exe</source>
            <destination>rebol.exe</destination>
        </file>
        <!--put any other data files here -->
    </files>
    <!-- $FINDEXE, $TMPRUN, $WINDIR, $PROGRAMDIR, $WINSYSDIR -->
    <onrun>$TMPRUN\rebol.exe -si $TMPRUN\script.r</onrun>
</xpackerdefinition>

Загрузите бесплатную программу XpackerX, измените вышеупомянутую матрицу так, чтобы
она содержала имя файла, которые вы дали своему сценарию, файлам поддержки, и
правильному пути к интерпретатору Rebol. Выполните XpackerX, и он создаст красиво
упакованный.exe файл, который не требует никакой инсталляции

14. Встраивание двоичных данных

Вы будете часто использовать изображения и другие двоичные данные в ваших программах
Rebol. Например, простые игры часто используют графические и звуковые файлы как
часть их интерфейса.  Вы можете читать их с жесткого диска, вы можете загрузить их
из Интернета, вы можете читать их из любого другого локального или сетевого носителя
данных, и т.д. Возможная альтернатива должна сделать файл, который включает вашу
программу и все ее вспомогательные файлы. Установка таких пакетов может быть сложной
для пользователей, различные операционные системы используют различные типы сжатия
(почтовый индекс, tar, и т.д. - и XpackerX  работает только в Windows). Чтобы
устранить эту проблему, Rebol предоставляет метод, чтобы закодировать и включить
внешние файлы в ваши программы. Чтобы видеть, как он работает, используйте код ниже:

Rebol [Title: "Конвертирование в двоичные данные"]

system/options/binary-base: 64
file: to-file request-file/only
data: read/binary file



editor data

Программа позволит вам выбирать файл, читать его, и затем отображать двоичное
представление его данных во встроенном редакторе. Вы можете скопировать эти данные и
вставить  непосредственно в вашу программу, назначить имя переменной, и использовать
эти данные, как будто они читаются прямо с жесткого диска или другого носителя
данных.

Вот пример. Загрузите изображение ниже, и затем сохранить образ на ваш жесткий диск.
Или, вы можете загрузить его, используя Rebol, как демонстрируется ранее в учебнике:

http://musiclessonz.com/test.png

Как только вы загрузили файл, используйте программу выше, чтобы выбрать и
конвертировать его в двоичные данные. Вы должны получить следующую распечатку:

64#{}

Теперь используем это:

picture: load 64#{}

Вы получите ответ ниже, что изображение определенно и создано:

make image! [85x100 #{
FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF
FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF...}

Здесь эта картинка находится в GUI:

view layout [image picture]

Пример ниже отображает фотографию моей собаки, и затем сохраняет ее на жесткий диск
как изображение.png:

Rebol []
dog: load 64#{

http://musiclessonz.com/test.png

gnY2kbYTCda2r0M36F9OXwq2qqlyqnyxPdk+5abS5z13LLMjFYu93j4eQt5lqCvI}

view layout [image dog]
save/png %dog.png dog

Код выше может быть сокращен, используя встроенное слово "compress". В Rebol
встроенное сжатие, и распаковка работают только с текстовыми строками. Таким
образом,  программа, предоставленная ранее (Конвертирование в двоичные данные)
должна быть откорректирована следующим образом:

REBOL [Title: "Конвертирование в двоичные данные"]

system/options/binary-base: 64
file: to-file request-file/only
if not file [quit]
uncompressed: read/binary file

compressed: compress to-string uncompressed
; код выше преобразовывает двоичные данные
; в текстовую строку, и затем сжимает ее.

editor compressed
alert rejoin ["Файл готов"]

Чтобы использовать сжатую версию данных выше, вы должны будете полностью изменить
текстово - двоичное преобразование после декомпрессации. Чтобы сделать это,
используйте следующий код:

to-binary decompress {compressed data}

Используя переменные выше, следующие две строки программы отображают то же самое
изображение:

view layout [загрузить распакованное изображение]
view layout [загрузить двоичное изображение для распаковки]

Вот законченный пример, демонстрирующий различие размера между исходником и сжатыми



данными:

; распакованное внедренное изображение:

image-uncompressed: load 64#{}

; Вот сжатая версия тех же данных:

image-compressed: load to-binary decompress 64#{}



view layout [не сжатые данные]
view layout [сжатые данные]

Когда вы используете сжатие как это, при загрузке убедитесь, что включали слово "to-
binary decompress".

 Вы можете преобразовать любой тип файла в сжатые текстовые данные. Изображения,
звуки, видео, даже все выполнимые программы могут быть включены в ваш исходный код!

15. Модульное программирование и повторное использование
кода

Многократное использование существующего кода является основным, если вы хотите
стать продуктивным программистом. Создание функций - основной способ осуществить
многократное использование кода. Как только вы создаете новое служебное слово, чтобы
достигнуть нужного действия, вы можете скопировать его определение и использовать
это много раз в ваших программах. Это избавляет вас от необходимости повторно
изобретать каждый раз функции и блоки, когда они необходимы. Также уменьшается
вероятность появления ошибок - старый код, не будет содержать ошибки, если это было
уже проверено в работе.

Вы можете импортировать существующие модули кода в вашу программу.  Существующий код
может содержать функциональные и переменные определения, новые программируемые
структуры, и даже завершить программы любой длины и уровня сложности. Как только
слова и определения были импортированы в вашу программу, вы можете использовать
включенные функции и переменные, как будто они - встроенные слова языка. Вам даже не
обязательно знать, как они были созданы. Наберите следующий пример кода, и сохраните
его к "C:\play_sound.r".

Rebol [title: "play-sound"]

play-sound: func [sound-file] [
wait 0

    ring: load sound-file
    sound-port: open sound://
    insert sound-port ring
    wait sound-port
    close sound-port
]

Код выше создает новое служебное слово "play-sound", который принимает переданное
переменное имя файла "sound-file" (файл должен быть ".wav"), и запускает звук через
компьютерные динамики. Теперь, всякий раз, когда  хотите запустить звук, вы можете
включить код в свою программу:

do %/c/play_sound.r

И используйте функцию "play-sound" точно так же как любое другое встроенное
служебное слово (синтаксис - "play-sound {sound-file}"):

play-sound %/C/WINDOWS/Media/chimes.wav

Чтобы подключить этот модуль в новую программу, надо ввести сроку
"%/c/play_sound.r". После этого определено слово "play-sound", и его можно
использовать на протяжении всей программы:

alert "Вот звук:"
play-sound %/C/1.wav



alert " И вот другой звук:"
play-sound %/C/2.wav

alert "Теперь попробуйте выбрать.wav файл на вашем жестком диске:"
play-sound to-file request-file/file %/C/1.wav

Чтобы снова использовать модули, необходимо сначала изучить язык, так, чтобы вы
могли понять код, написанный другими разработчиками. Вы, возможно, должны
приспособить и расширить код, написанный другими, так, чтобы он соответствовал вашим
потребностям более точно. В большинстве случаев, однако, вы можете только узнать,
как использовать слова в импортированном модуле, или диалекте, выполнить "do
{filename}" команду.

Вебсайт http://www.dobeash.com/it/ предлагает несколько свободных модулей расширения
для языка Rebol. Например, модуль "RebGUI", расширяет встроенный в Rebol уже сильный
синтаксис GUI, встроенный в Rebol ("view layout..."). Используя RebGUI, можно легко
создать графические элементы, которые не возможны в Rebol. Чтобы использовать
RebGUI,  загрузите файлы с вебсайта выше, и распакуйте на диске C:\. . Вы можете
подключить, это любым из средств, описанных в предыдущей главе ("do
%/c/rebgui_example.r", и т.д.):

REBOL []

do %\c\rebgui.r

display "Сетка" [
table #WH options [

        "День" left .5 "Время" left .3 "Имя" left .3
    ] data [
        Monday 9:00 "Вася" Tuesday 9:30 "Маша" Wednesday 10:00 "Миша"
    ]
]

do-events

Код выше использует некоторые новые команды, которые не являются встроенной частью
языка Rebol. Функции и переменные "display", "table", "#WH", "options", и "data"
определены в файле rebgui.r, и они добавляют новые функциональные возможности к
языку Rebol. Используя те слова, как определено в rebgui.r, вышеупомянутый код
показывает данный блок данных [Monday 9:00 "Вася Tuesday" 9:30 "Маша Wednesday"
10:00 "Миша"] в GUI, где можно автоматически сортировать, нажимая на заголовки в
колонках GUI. Этот тип интерфейса - общее требование в современных программах,
которые показывают списки данных. RebGUI содержит широкую коллекцию дополнительных
функций, которые полезны в построении сложного GUI. Чтобы использовать их, вы должны
сначала понять основной синтаксис Rebol, и затем узнать, как использовать языковые
расширения RebGUI. Как только вы освоите это, вы можете просто включить файл
"rebgui.r"  в свои программы, и использовать эти языковые расширения, как будто они
- часть языка. Чтобы их использовать, вы должны сделать импорт rebgui.r.

Вот некоторые сайты, содержащие свободные модули, которые могут быть  вам полезными:

http://www.hmkdesign.dk/rebol/list-view/list-view.r - Возможно это, самое полезное
дополнение к Rebol GUI.

http://www.dobeash.com/it/rebdb/ - модуль базы данных, который позволяет вам легко
хранить и организовывать большое количество данных, используя язык данных "SQL".
Есть также модуль спеллчекера, который может быть включен в ваши программы.

http://www.rebol.org/cgi-bin/cgiwrap/rebol/view-script.r?script=rebzip.r - модуль
для сжатия и распаковки zip файлов.

http://www.colellachiara.com/soft/Misc/pdf-maker.r - диалект для создания pdf

http://www.hmkdesign.dk/rebol/list-view/list-view.r
http://www.dobeash.com/it/rebdb/
http://www.rebol.org/cgi-bin/cgiwrap/rebol/view-script.r?script=rebzip.r
http://www.colellachiara.com/soft/Misc/pdf-maker.r


файлов.

http://softinnov.org/rebol/mysql.shtml - модуль, чтобы управлять mysql базами данных
в пределах Rebol. Очень полезный для веб-программирования.

http://www.rebol.org/cgi-bin/cgiwrap/rebol/view-script.r?script=menu-system.r -
диалект, чтобы создать все типы меню GUI в Rebol. Более поздняя глава посвящена этой
теме.

http://softinnov.org/rebol/uniserve.shtml - структура, чтобы помочь построить
сетевые приложения клиент-сервер.

http://www.rebol.net/demos/BF02D682713522AA/i-rebot.r
http://www.rebol.net/demos/BF02D682713522AA/objective.r
http://www.rebol.net/demos/BF02D682713522AA/histogram.r - эти примеры содержат
трехмерный машинный модуль, написанный полностью на чистом Rebol. Модуль позволяет
вам легко добавлять и управлять трехмерными графическими объектами в ваших
приложениях Rebol.

http://web.archive.org/web/20030411094732/www3.sympatico.ca/gavin.mckenzie/ -
библиотека анализатора (парсер) Rebol XML.

http://box.lebeda.ws/~hmm/rswf/ - диалект, чтобы создать flash (SWF) файлы
непосредственно из скриптов Rebol.

http://www.rebol.net/docs/makedoc.html - конвертирование текстовых файлов в
отформатированные файлы HTML.

http://www.rebol.org - вся библиотека Rebol - полная многих дополнительных модулей и
полезных кодовых фрагментов. Первое место, где нужно искать необходимый исходный
текст Rebol.

Используйте внешние программы как "modules":

Помните, что данные часто являются общими между существующими программами.
Встроенное слово Rebol "Call " позволяет вам управлять другими программами на вашем
компьютере.  Используя "Call", вы можете выполнить все команды "shell", включенных в
операционную систему вашего компьютера (то есть, DOS и команды Unix). Пример ниже
открывает, блокнот Windows, чтобы отредактировать "rebgui_example.r":

call "notepad.exe c:\rebgui_example.r"

Этот следующий пример открывает программу Paint, чтобы отредактировать изображение,
которое мы загружали ранее:

call "mspaint.exe c:\bay.jpg"

Вот пример, который включает выполнимую программу в код, развертывает, и пишет
программу накопителю, и затем управляет этим с функцией запроса:

program: load to-binary decompress 64#{http://softinnov.org/rebol/mysql.shtml
http://www.rebol.org/cgi-bin/cgiwrap/rebol/view-script.r?script=menu-system.r
http://softinnov.org/rebol/uniserve.shtml
http://www.rebol.net/demos/BF02D682713522AA/i-rebot.r
http://www.rebol.net/demos/BF02D682713522AA/objective.r
http://www.rebol.net/demos/BF02D682713522AA/histogram.r
http://web.archive.org/web/20030411094732/www3.sympatico.ca/gavin.mckenzie/
http://box.lebeda.ws/%7Ehmm/rswf/
http://www.rebol.net/docs/makedoc.html
http://www.rebol.org/

}
write/binary %program.exe program
call %program.exe

16. Ключевые характеристики Rebol

Список ниже показывает некоторые ключевые особенности языка Rebol. Знание, как
использовать эти элементы, составляет фундаментальное понимание того, как Rebol
работает:

1.  Для начала, в Rebol есть сотни встроенных служебных слов, которые выполняют
общие задачи. Как и в других языках, служебные слова типично сопровождаются
переданными параметрами. Чтобы достигнуть цели, функции устроены так чтобы,
один за другим закончить данную задачу. Выражения идут слева направо, отступы
строк и пробелы не требуются, но могут быть вставлены, чтобы сделать код
читаемым. Текст после точки с запятой и перед новой строкой рассматривают как
комментарий. Только знание предопределенных функций в языке и способность
использовать их в полной мере позволит вам работать с Rebol.

2. Много общих типов данных понятны интерпретатору Rebol. Числа, текстовые
строки, деньги, время, кортежи, URL, двоичные данные изображений, звуков, и
т.д. К ресурсам сети и интернет-протоколам (http документы, справочники
программы передачи файлов, электронной почты, dns услуги, и т.д.) можно легко
получить доступ. Данные любого типа могут быть написаны и прочитаны с
фактически любого связанного устройства или ресурса. Помните, что символ
процента ("%") используется, чтобы обратиться к файлам.

3. И код управления и данные сохранены в "блоках", заключенные скобки ("[]").
Данные и код, содержавшийся в блоках, отделены пробелом. Функции,
содержавшиеся в блоках, могут быть оценены (их выполненные действия), с
помощью слова "do". Новые служебные слова могут быть созданы, используя
"func". Слово "func" позволяет вам передавать свои собственные указанные
параметры к недавно определенным служебным словам. Вы можете использовать
модуль кода, содержавшегося в текстовом файле, пока он содержит минимальный
заголовок "rebol []". Данные, содержавшиеся в блоках, могут иметь любого типа,
и блоки можно автоматически рассматривать как списки данных, названных
"series". Последовательностью можно управлять, используя встроенные функции,
которые позволяют поиск, сортировку, и иную организацию данных собранных в
блоках.

4. Синтаксис "view layout [block]" используется, чтобы создать базовые GUI.
Добавьте слова к блоку, чтобы создать графические элементы: "button", "field",
"text-list", и т.д. Добавьте цвет, положение, интервал, и другие слова аспекта
после каждого слова элемента. Добавьте блоки действия для создания активного
элемента. Обратитесь к пунктам в расположении GUI, используя обработки пути
(то есть, "face/offset" обращается к положению элемента). Эти простые
руководства могут использоваться, чтобы создать хороший GUI.

5. Любым данным или программному коду, можно назначить слова как имена. Имена
назначаются на значения, переменные, функции, выражения, и блоки любого типа,
используя двоеточия (":"). После этого назначенные переменные слова могут
использоваться, чтобы представить все функции и данные, содержавшиеся в данном



выражении, блоке, и т.д. Поместите двоеточие в конце слова, и после это
передавайте все следующие действия или данные. Это является базовой частью
языковой структуры Rebol.

17.Завершенные Rebol программы для исследований.

Следующие программы используют множество методов, о которых говорилось выше. Они
демонстрируют, как части кода могут быть соединены, чтобы создать законченные
приложения, и раскрывают, как могут быть собраны основные стандартные блоки языка в
работающую программу.  Изучение этих примеров - возможно, самая ценная часть этого
обучающего материала.

Загружаемый пакет всех демонстрационных программ доступен в:
http://musiclessonz.com/rebol_tutorial_examples.zip. Файл содержит скриншот,
отделенный исходный текст, и "exe" файлы каждого примера. XpackerX XML файлы, также
включены.

mystring: trim {Это строка}

Тоже самое:

mystring: "Это строка"

"Join" использовался, чтобы присоединить слово к длинным разделам текста:

join "Всё " [
    "равно "
    "линия "
    "текста."
]

тоже самое:

"Всё равно линия текста."

Блоки могут быть написаны с переносом строк:

myblock: [Это
блок
]

тоже самое:

myblock: [Это тоже блок]

Чтобы выполнить примеры программ из этого раздела, вы можете использовать любой из
методов, описанных ранее: наберите код в Rebol интерпретатор, сохраните программу
как текстовый файл и выполните его, используя "do {filename}", или упакуйте
XpackerX, и т.д.

17.1 Маленький почтовый клиент.

Первый пример - почтовый клиент, который может использоваться, чтобы читать и



посылать сообщения. Код полностью прокомментирован, чтобы обеспечить объяснения
каждой строки, и как каждый элемент работает:

Rebol [Title: "Маленький почтовый клиент"]
; (каждая программа требует минимального заголовка)

view layout [
        ; строка выше создает GUI

h1 "Отправить почту:"
        ; Эта строка добавляет текстовую метку к GUI

address: field "recipient@website.com"
        ; Эта линия создает текстовое поле для ввода, содержащее
        ; заданный по умолчанию адрес "recipient@website.com". Это назначает
        ; переменное слово "address" к тексту, введенному здесь.
    subject: field "Тема"

; другое текстовое поле для ввода темы письма
body: area "Здесь ваше послание"

        ; Это создает большую, многострочную текстовую область ввода для
        ; основного текста послания.

btn "send" [
            ; Создана кнопка для отправки писем.

send/subject to-email address/text body/text subject/text
            ; Использует встроенное слово "send", чтобы послать
            ; электронную почту.
            ; посылает функцию, с ее "/subject" на обработку
            ; принимая три параметра. Это передает
            ; текст, содержавшийся в области, показано выше(названный
            ; "address/text" "body/text" и
            ; "subject/text"). Встроенная функция "to-email"
            ; гарантирует, что текст адреса рассматривают как
            ; данные электронной почты.
        alert "Сообщение отправленно."

; приводит пользователя в готовность, когда предыдущая строка закончена.
    ]

h1 "Читать почту:"
; Другая текстовая метка

mailbox: field "pop://user:pass@website.com"
; Другое текстовое поле ввода. e-mail пользователя

        ; информация введена здесь.
    btn "Читать" [

; Дополнительная кнопка, на сей раз с действующим
            ; блоком, который читает сообщения от указанного почтового ящика.

editor read to-url mailbox/text
            ; Встроенная функция "to-url" гарантирует что
            ; текст в поле почтового ящика обработан как url.
            ; Информационные наполнения почтового ящика читаются и отображены
            ; во встроенном редакторе Rebol.
    ]
]

Вот тот же самый код, без комментариев - это очень просто:

Rebol [Title: "Маленький почтовый клиент"]

view layout [
h1 "Отправить почту:"
address: field "recipient@website.com"

    subject: field "Тема"
body: area "Здесь ваше послание"

    btn "send" [
send/subject to-email address/text body/text subject/text

        alert "Сообщение отправлено."
   ]

h1 "Читать почту:"
mailbox: field "pop://user:pass@website.com"



    btn "Читать" [
editor read to-url mailbox/text

    ]
]

17.2 FTP чат

Второй пример - простое приложение чата, которое позволяет пользователям
обмениваться сообщениями через Интернет. Это включает защищённый паролем доступ для
администраторов, чтобы стереть информационные наполнения чата. Это также позволяет
пользователям приостанавливать деятельность, и запрашивать продолжения через имя
пользователя и пароль ["secret" "password"]. Чат создан для, динамического создания,
чтения, и сохранения текстовых файлы через FTP (чтобы использовать программу, вы
должны будете обратиться к ftp-серверу: адрес программы передачи файлов, имя
пользователя, и пароль).

Rebol [title: "FTP чат"] ; необходимый заголовок

webserver: to-url request-text/title/default trim {
Адрес веб-сервера:} {ftp://user:pass@website.com/chat.txt}

; Имя пользователя программы передачи файлов, пароль, домен, и
; имя файла должны быть введены в форму.

name: request-text/title "Введите имя:"
cls: does [prin "^(1B)[J"]
; "cls" теперь очищает экран (объясненный ранее ).

write/append webserver join now [
": " name " вошел." newline

]
; Строка выше пишет некоторый текст на web-server.
; "/append", добавляет к существующему файлу
; на web-server  (в противоположность стиранию).
; Используя "join", записывает на web-server - объединенное
; значение, {имя пользователя}, некоторый статический текст,
; дату и время.
forever [

current-chat: read webserver
    ; читает сообщения, которые в настоящее время находятся на web-server,
    ; и назначает переменной слово "current-chat"

cls ; очищает экран, используя слово, определенное выше
    print join "--------------------------------------------------" [
        newline { Вы вошли как: } name newline
        {Type "участок памяти" выбрать место для дискуссий.} newline
        {Type "блокировка" приостановить или блокировать ваш чат.} newline
        {Type "выход" закончить ваш чат.} newline
        {Type "очистить" удалить текущий чат.} newline
        {Press [ENTER] периодически обновлять дисплей.} newline
        "--------------------------------------------------" newline]
    ; отображает приветствие и некоторые команды

print join " Текущий текст в чате: " [webserver newline]
print current-chat

    sent-message: copy join name [
        " Вам говорят: " entered-text: ask " Вы говорите:  "
    ]

switch/default entered-text [
        "выход"  [break]
                ; если пользователь выбрал "выход",
                ; остановит цикл с выходом из прогаммы
        "очистить" [

if/else request-pass = ["пароль"] [
write webserver ""] [alert trim {
Вы должны ввести пароль администратора
чтобы очистить участок памяти!}



            ]
        ]
                ; если пользователь выбрал "очистить", для стирания
                ; текущего текста в чате
                ; спросит пароль администратора.

        "участок памяти"  [
write/append webserver join now [

                ": " name " вышел." newline]
webserver: to-url request-text/title/default {Новый адрес веб-сервера:}

to-string webserver
            write/append webserver join now [
                ": " name " вошел." newline
            ]
        ]
        "блокировка" [

alert trim { Программа сделает паузу в течение 5 секунд.
                Чтобы продолжить, необходимы правильные имя и пароль.}
            pause-time: now/time + 5
                        forever [if now/time = pause-time [

; ждем 5 секунд
                while [request-pass <> ["пароль"]] [

alert "Пароль неверный – доступ запрещен!"
                    ]
                break
                ]
            ]

; выйдет из программы после 5 секунд.
        ]
    ] [if entered-text <> "" [
            write/append webserver join sent-message [newline]]]
]
; при выходе из цикла "forever" из цикла выходят, сделаем следующее:
cls print "Досвидания!"
write/append webserver join now [
    ": " name " чат закрыт." newline]
ждите 1

Объем программы выполняется в пределах цикла "forever", и использует условный
оператор "switch", чтобы решить, как ответить на пользовательский ввод. Это -
классическая структура, которая может быть откорректирована, чтобы соответствовать
множеству обобщенных ситуаций, в которых компьютер неоднократно ждет и отвечает на
пользовательские запросы.

17.3 Использование цикла для обработки данных

Одно из самых важных приложений структур цикла должно пройти через списки данных.
Шагая через элементы в блоке, циклы могут использоваться, чтобы обработать и
выполнить действия для каждого элемента данных. Эта методика используется во всех
типах программирования, и это - краеугольный камень способа, которым программисты
пользуются в работе с таблицами данных. Поскольку много программ работает со
списками данных, вы будете очень часто сталкиваться с ситуациями, которые требуют
использования циклов. Пример ниже демонстрирует несколько путей, которыми вы будете
пользоваться.

; Во-первых, маленькая пользовательская база данных определена. Это организовано
; в блочную конструкцию: "users" блок содержит 5
; блоков, для которых каждый содержит 5 элементов информации
; каждого пользователя. Пропуски в анкете представлены с пустыми котировками.
; Отметьте, что каждый блок разбит на две строки, так, чтобы
; они поместились на эту страницу (они не должны быть разбиты ):

users: [
    ["Саша" "Петя" "Москва"
        "Лубянка" "555-1234"]
    ["Паша" "Толик" "Новгород"
        "Заводская" "555-2345"]



    ["Маша" "Наташа" "Омск"
        "Набережная" "555-3456"]
    ["Гриша" "Миша" "Томск"
        "Советская" ""]
    ["Ира" "Соня" ""
        "" "555-5678"]
]

;  Эта программа не имеет GUI. Основа этого, консольная программа.

draw-line: does [loop 65 [prin "-"] print ""]

; Это не самый эффективный способ, ниже пример лучше.
; a-line: copy []
; loop 65 [append a-line "-"]
; ; remove the spaces and turn it
; ; into a string of characters:
; a-line: trim to-string a-line
; ; now you can print "a-line"
; ; anywhere you need it:
; print a-line
;
; Неэффективный код выше остается в этом примере чтобы
; показать, разницу работы с эффективным кодом.

; Следующая - маленькая функция, которая распечатывает все данные
; в базе данных. Это использует foreach цикл, чтобы циклически пройти
; каждый блок пользовательских данных, и затем печатает
; отображение каждого элемента в блоке.
print-all: does [

foreach user users [
draw-line

        print rejoin ["Пользователь:     " user /1 " " user /2]
        draw-line

print rejoin ["Адресс:  " user /3 "  " user /4]
print rejoin ["Телефон:    " user/5]
print newline

    ]
]

; Следующий код непрерывно использует цикл forever,
; запрашивает выбор у пользователя. Использует несколько
; циклов foreach, чтобы выбрать информацию из базы данных, и
; условную структуру "switch", чтобы решить, как ответить
; на запрос пользователя:

forever [

prin "^(1B)[J" ; этот код очищает экран.
print " Вот текущие пользователи в базе данных:^/"

    ; "^ /" в конце строки для перехода на новую строку.
draw-line

    ; Теперь выводит список имен пользователей. foreach цикл
    ; используется, чтобы получить первое и последнее имя каждого
    ; пользователя в базе данных.

foreach user users [prin rejoin [user/1 " " user/2 "  "]]
print "" draw-line

    ; Теперь спросите у пользователя выбор:

print " Введите имя пользователя ниже.^/"
    print " Выберите 'all' для законченного листинга базы данных."

print "Нажмите [Enter] для выхода.^/"
answer: ask { О каком человеке нужна информация? }
print newline

; Теперь решите, что сделать с ответом пользователя:



switch/default answer [
        ; Если они выбрали "all", выполните "print-all"
        ; функция определила ранее:

        "all" [print-all]

; Если нажата только [Enter] (""), выводит
        ; прощальное сообщение, и заканчивает программу.

; "ask" используется, чтобы отобразить сообщение, вместо
        ; "print". Это позволяет программе ждать пока пользователь
        ; не нажмет клавишу прежде, чем закончить программу:

"" [ask "До свидания!  Нажмите [Enter] для выхода." quit]
        ; Если ни один из вариантов выше не был выбран,
        ; ниже выполняется блок значения по умолчанию
        ; (последняя часть структуры выключателя):

        ][

        ; Этот раздел начинается, с создания переменной "flag",
        ; которая используется, чтобы проследить существует ли
        ; пользователь в базе данных.
        ; Слово "found" показывает что имя пользователя еще не было найдено:

        found: false

; Затем, цикл foreach цикл шагает через каждый
; пользовательский блок в базе данных:

foreach user users [

            ; Если введенное имя пользователя найдено в базе данных (или
; первое или фамилия), информация этого пользователя выводится

            ; на экран.

if find rejoin [user/1 " " user/2] answer [
draw-line
print rejoin ["Пользователь:     " user/1 " " user/2]
draw-line

                print rejoin ["Адресс:  " user/3 " " user/4]
print rejoin ["Телефон:    " user/5]
print newline

                found: true
            ]
        ]

        ; Если переменная "found"  - все еще ложь после выполнения
        ; цикла через всю пользовательскую базу данных, и имя
        ; пользователя не было найдено в базе данных.
        ; Выведем сообщение на экран:

        if found <> true [
print "Данных нет!^/"]

    ]
ask "Нажмите [ENTER] для выхода"

]

17.4 Эффекты для изображений

Следующее приложение создает GUI, загружает и отображает изображение из Интернета,
позволяет применять эффекты к нему, и сохранять на жесткий диск.

REBOL [Title: ""]

effect-types: ["Invert" "Grayscale" "Emboss" "Blur" "Sharpen"
                "Flip 1x1" "Rotate 90" "Tint 83" "Contrast 66"
                "Luma 150" "None"]



; создает список эффектов, которые встроены в Rebol, и
; назначены переменному слову "effect-types" в блоке

do %/c/play_sound.r
; Строка выше импортирует простую функцию "play-sound"
; созданную ранее. Чтобы программа работала
; правильно файл play_sound.r  должен находится на C:\

image-url: to-url request-text/title/default {
Введите url откуда использовать изображение:} trim {
http://rebol.com/view/demos/palms.jpg}

; спрашивает у пользователя расположение изображения (со
; значением по умолчанию местоположения), и назначает это всё на слово "new-image"

gui: [
; Следующий код отображает меню программы, используя
; "choice" графический фрагмент кнопки (тип выбора меню
; встроенный к Rebol). Кнопка составляет 160 пикселей , и помещена в верхний,
крайний ; левый угол в GUI (0x0) используя встроенное слово "at".

    across
; горизонтально выравнивает все следующие графические
; фрагменты GUI.

    at 20x2 choice 160 tan trim {
Сохранить изображение} "Показать сохраненное изображение" "Скачать новое"

trim {
-------------} "Выход" [

        if value = "Сохранить изображение" [
filename: to-file request-file/title/file/save trim {

Сохранить как:} "Сохранить" %/c/effectedimage.png
           ; просит как сохранить файл,
           ; по умолчанию "c:\effectedimage.png"

save/png filename to-image picture
            ; сохраняет на жесткий диск
        ]

if value = "Показать сохраненное изображение" [
            view-filename: to-file request-file/title/file trim {

Сохранить файл:} "Сохранить" filename
            view/new center-face layout [image load view-filename]
            ; читает выбранное изображение с жесткого диска
            ; и отображает это в новом окне GUI
        ]

if value = "Скачать новое изображение" [
new-image: load to-url request-text/title/default trim {

                Введите url нового изображения} trim {
http://www.rebol.com/view/bay.jpg}

            ; просит местоположение нового изображения,
            ; и назначает это на слово "new-image"

picture/image: new-image
; заменяет старое изображение новым
show picture ; обновляет экран GUI

        ]
if value = "-------------" [] ; ничего не делает
if value = "Exit" [

            play-sound %/c/windows/media/tada.wav
quit ; выход из программы

        ]
    ]

    choice tan "Информация" "О программе"
        [alert "Эффекты изображения - Copyright 2005, Nick Antonaccio"]
    ; простой блок "О программе"

    below
; вертикально выравнивает фрагменты виджета

    ; противоположность "across"

    space 5
; некоторые фрагменты виджета



    pad 2
; помещает 2 пикселя пустого пространства перед виджетом

    box 550x1 white
; чертит линию 550 пикселей шириной, 1 пиксель высотой
; (декоративный разделитель)

    pad 10
; помещает некоторое пространство между виджетами

    vh1 " Двойной щелчок на каждом эффекте в списке справа:"
; большой заголовок текста для GUI

    return
    ; усовершенствования к следующей строке в GUI
    across

    picture: image load image-url
    ; вводит изображение в начале программы,
    ; и дает ему метку

    text-list data effect-types [
current-effect: to-string value

        picture/effect: to-block form current-effect
        show picture
    ]

    ; Код выше создает виджет текстового списка и назначает блок
    ; действий к нему, которые выполняются всякий раз, когда
    ; пользователь нажимает на список. Блок действий выровнен
    ; и каждое действие помещено в отдельную строку для легкости
    ; прочтения кода.
    ; Первая строка назначает слову "current-effect" значение
    ; которое пользователь выбрал от списка. Вторая строка
    ; применяет этот эффект к изображению.
    ; Третья строка отображает недавно обработанное изображение.
]

view/options center-face layout gui [no-title]

17.5 Игра мозаика

Вот простой пример, который осуществляет GUI версию классической мозаичной игры в 8
строках кода. Поле содержит 15 подвижных частей и 1 пустое пространство.

Rebol [Title: "Мозаика"]

gui: [

    ; Определите основные параметры схемы размещения. "origin 0x0"
    ; запускает схему размещения в верхнем левом углу GUI.
    ; "across" располагает рядом друг с другом виджеты:

    origin 0x0 space 0x0 across

    ; Слово "style" ниже позволяет вам переопределять
    ; появление и характеристики любого действия  встроенного в
    ; виджет. Раздел ниже создает недавно определенный стиль
    ; кнопки "piece", с блоком действий, которые меняют позицию
    ; текущей кнопки. Эти действия работают всякий раз, когда одна
    ; из кнопок нажата.

style piece button 60x60 [

; Строка ниже выясняет, нажата кнопка смежено с пустым
        ; пространством или нет. "offset" содержит позицию данного
        ; графический фрагмент. Слово "face" используется, чтобы
        ; обратиться к нажимаемому в настоящее время фрагменту.

        if not find [0x60 60x0 0x-60 -60x0]
face/offset - empty/offset [exit]



temp: face/offset

face/offset: empty/offset

empty/offset: temp
    ]

; Строки ниже располагает кнопки стиля "piece" на
    ; экран GUI.

    piece "1" piece "2" piece "3" piece "4" return
piece "5" piece "6" piece "7" piece "8" return

    piece "9" piece "10" piece "11" piece "12" return
    piece "13" piece "14" piece "15"
    empty: piece 200.200.200 edge [size: 0]
]

; Отображает целый блок GUI:

view center-face layout gui

17.6 Гитарные аккорды

Пятый пример - программа, которая создает, сохраняет, и печатает коллекции гитарных
аккордов. Это демонстрирует полезный пример, и методику манипуляции GUI:

Rebol [Title: "Гитарные аккорды"]

; загрузим внедренные изображения:

fretboard: load 64#{
iVBORw0KGgoAAAANSUhEUgAAAFUAAABkCAIAAAB4sesFAAAACXBIWXMAAAsTAAAL
EwEAmpwYAAAA2UlEQVR4nO3YQQqDQBAF0XTIwXtuNjfrLITs0rowGqbqbRWxEEL+
RFU9wJ53v8DN7Gezn81+NvvZXv3liLjmPX6n/4NL//72s9l/QGbWd5m53dbc8/kR
uv5RJ/QvzH42+9nsZ7OfzX62nfOPzZzzyNUxxh8+qhfVHo94/rM49y+b/Wz2s9nP
Zj+b/WzuX/cvmfuXzX42+9nsZ7OfzX4296/7l8z9y2Y/m/1s9rPZz2Y/m/vX/Uvm
/mWzn81+NvvZ7Gezn8396/4l2/n+y6N/f/vZ7Gezn81+tjenRWXD3TC8nAAAAABJ
RU5ErkJggg==
}

barimage: load 64#{
iVBORw0KGgoAAAANSUhEUgAAAEoAAAAFCAIAAABtvO2fAAAACXBIWXMAAAsTAAAL
EwEAmpwYAAAAHElEQVR4nGNsaGhgGL6AaaAdQFsw6r2hDIa59wCf/AGKgzU3RwAA
AABJRU5ErkJggg==
}

dot: load 64#{
iVBORw0KGgoAAAANSUhEUgAAAAoAAAAKCAIAAAACUFjqAAAACXBIWXMAAAsTAAAL
EwEAmpwYAAAAFElEQVR4nGNsaGhgwA2Y8MiNYGkA22EBlPG3fjQAAAAASUVORK5C
YII=
}

; Дизайн Gui:

; Подпрограмма ниже была скопирована с
; http://rebol.com/how-to/feel.html

movestyle: [
engage: func [face action event] [

if action = 'down [
face/data: event/offset

            remove find face/parent-face/pane face
            append face/parent-face/pane face
        ]

if find [over away] action [
face/offset: face/offset + event/offset - face/data



        ]
show face

    ]
]

gui: [
backdrop white

        ; делает белый фон GUI
    currentfretboard: image fretboard 255x300

currentbar: image barimage 240x15 feel movestyle
text "КОМАНДЫ:" underline
text "Перетащите точки и другие графические фрагменты на поле."
across
text "Измените размеры поля:"
tab
rotary "255x300" "170x200" "85x100" [

currentfretboard/size: to-pair value show currentfretboard
        switch value [
            "255x300" [currentbar/size: 240x15 show currentbar]
            "170x200" [currentbar/size: 160x10 show currentbar]
            "85x100" [currentbar/size: 80x5 show currentbar]
        ]
    ]

    return
    button "Сохранить" [

filename: to-file request-file/save/file "1.png"
save/png filename to-image currentfretboard

    ]

    tab

    button "Печать" [
        filelist:
            sort request-file/title "Выберите изображение, чтобы напечатать:"

html: copy "<html><body>"
foreach file filelist [

append html rejoin [
{<img src="file:///} to-local-file file {">}

            ]
        ]
        append html [</body></html>]

write %chords.html trim/auto html
        browse %chords.html
    ]
]

loop 50 [append gui [at 275x50 image dot 30x30 feel movestyle]]
loop 50 [append gui [at 275x100 image dot 20x20 feel movestyle]]
loop 50 [append gui [at 275x140 image dot 10x10 feel movestyle]]

loop 6 [append gui [at 273x165 text "X" bold feel movestyle]]
loop 6 [append gui [at 273x185 text "O" bold feel movestyle]]

view layout gui

17.7 База данных

Этот пример - простая программа базы данных с GUI. Сохраняет в память, ищет,
изменяет, организовывает и отображает данные. Этот тип приложения распространен в
любой ситуации, где нужная информация должна быть организована. Хотя этот пример
обрабатывает клиентскую информацию списка, методики, которые здесь демонстрируются,
могут быть непосредственно применены к другим типам приложений - материальные
ценности, списки клиентов и продавцов, отчеты платежной ведомости и налога,
мультимедийные коллекции, альбомы, фотографии, и т.д. Эта программа использует
listview модуль, найденный в http://www.hmkdesign.dk/rebol/list-view/list-view.r,
который сжат и находится в пределах скрипта. listview модуль делает всю основную
работу отображения, сортировки, фильтрации, изменения, и управления данными, со



знакомым пользовательским интерфейсом. Документация доступна в
http://www.hmkdesign.dk/rebol/list-view/list-view.html. Изучение использования
listview является полезным, если вы намереваетесь сделать программный продукт,
который использует информационное управление.

Программа ниже создает заданную по умолчанию базу данных. Есть дополнительная
функция, чтобы препятствовать, программе закрываться случайно.

Rebol [title: "Моя база данных"]

evt-close: func [face event] [
    either event/type = 'close [

inform layout [
across
Button "Сохранить" [

                ; при нажатии кнопки, резервные данные сохраняются
                ; в автоматически созданном файле:

backup-file: to-file rejoin ["backup_" now/date]
                write backup-file read %database.db
                save %database.db theview/data quit
            ]

Button "Не сохранять" [quit]
Button "Отмена" [hide-popup]

        ] none ] [
event

    ]
]
insert-event-func :evt-close

; Код ниже - модуль list-view.r в сжатом формате.

do decompress #{
789CCD3D6B73E3C871DFF52BE6369592F67629905A9F73A6B2A7B2EFECD8E557
2AE738A96221551031142181040D8092B8A9FC97FCD474F7BCBA07038ADA3B3B
E6D56909CCABA7BBA75FD333FCB75FFEE28FBF538B33A5FE54F5B59EAB377FFE
CD77EA77BFF9FE4F933FFFE697FFA17E552CF51B28FD558585FF58575D3F79AC
F4D3650B2F7FBEEFD74D3B578B37BFD6DBB67A50BFAF1E0A5DABDFB6504D6F3B
BD7D9343B56F9BDDA1ADEED63DF47E359D7EF55EC1DF9FAA89FAF5EF7FABBED3
AEB0EC0ACB2A3024617DF7601201404537C1CEA18383A79B6A197A56FBBAB6540E08D62033C180F121AD8C7DAD0CEC608DE22EAEAD3EA82B8FEF0017D79CC0DD662E0A434C0A3E1AA49033C0792DC76E897A2E5B5F1CAA4A9E759747CC44963FA55A3CCF7CC10B6437831D0409ACA1E345FAE7CC1C835AE039ACA3BC1D4F182D}

; Следующая условная оценка выясняет какие файл базы данных
; существуют. В противном случае создает файл с пустым блоком

if not exists? %database.db [write %database.db {[][]}]

; Теперь хранящиеся данные назначаются в переменное слово:

database: load %database.db

view center-face gui: layout [
h4 { Чтобы ввести данные, дважды щелкните любую строку, и набирайте

непосредственно в колонках.}
theview: list-view 800x200 with [

data-columns: [Студент Учитель День Время Телефон
            Родители Возраст Платежи Переупорядочить Примечания]

data: copy database
        tri-state-sort: false
        editable?: true
    ]

across
button "+ строка" [theview/insert-row]
button "- строка" [

if (to-string request-list "Правильно?"
                [да нет]) = "да" [

theview/remove-row
        ]
    ]
    button "фильтр" [

filter-text: request-text/title trim {
            Фильтр (оставьте пробел, чтобы обновить все данные):}

if filter-text <> none [
theview/filter-string: filter-text

            theview/update
        ]
    ]

button "сохранить db" [
backup-file: to-file rejoin ["backup_" now/date]
write backup-file read %database.db

        save %database.db theview/data
    ]
]

Ниже упрощенная версия программы “Моя база данных” без встроенного list-view.r:

Rebol []

if not exists? %list-view.r [write %list-view.r read
    http://www.hmkdesign.dk/rebol/list-view/list-view.r
]
do %list-view.r
if not exists? %database.db [write %database.db {[][]}]
database: load %database.db



view center-face gui: layout [
theview: list-view 800x200 with [

data-columns: [Студент Учитель День Время Телефон
            Родители Возраст Платежи Переупорядочить Примечания]

data: copy database
        tri-state-sort: false
        editable?: true
    ]

across
button "+ строка " [theview/insert-row]

    button "- строка " [theview/remove-row]
    button " фильтр" [

filter-text: request-text/title trim {
Фильтр (оставьте пробел, чтобы обновить все данные):}

if filter-text <> none [
theview/filter-string: filter-text

            theview/update
        ]
    ]
    button " сохранить db " [save %database.db theview/data]
]

Особенности манипуляции данных в примере выше допускаются импортированным модулем
list-view. Это делает создание и управление базами данных, в Rebol достаточно легкой
процедурой. Используем схему выше как основу, чтобы создавать графический интерфейс
для любого типа Rebol приложения базы данных. Это отличный инструмент, чтобы начать
разрабатывать программы, и работать с данными в любой сфере, где это необходимо.

Для получения дополнительной информации о том, как работать со списками данных,
используя инструментальные средства Rebol, см. страницы ниже:

http://compkarori.com/vanilla/display/VID+list+style

http://compkarori.com/vanilla/display/list+examples

Ссылки ниже для примеров того, как организовать, и управляли данными, используя
чистый синтаксис Rebol, и некоторые сторонние модули:

http://www.rebol.net/cookbook/recipes/0012.html

http://www.dobeash.com/it/rebdb/

http://softinnov.org/rebol/mysql.shtml

Есть ссылки к инструментальным средствам базы данных на rebDB странице. Понимание и
способность использовать таких инструментальных средств - важная часть создания
современных приложений всех типов.

17.8 Одноранговый сетевой менеджер

Этот пример позволяет двум пользователям соединяться непосредственно через порт сети
TCP/IP, для обмена сообщениями. В отличие от чата FTP выше, этот пример не требует,
чтобы сервер  хранил или передавал данные. Текст посылают непосредственно между
двумя компьютерами, через установленное сетевое подключение (в Интернете или
локальной сети). Приложение может действовать как клиент или как сервер, в
зависимости от выбора пользователя. Чтобы работать должным образом, у машины сервера
должны быть выставленный адрес IP или открытый порт маршрутизатора и межсетевой
защиты. Клиентская машина может быть расположена позади маршрутизатора или
межсетевой защиты, без любых отправленных входящих портов. Операция программы может
быть продемонстрирована на единственном компьютере. Для команд, см. документацию
справки, включенную в код.

http://www.rebol.net/cookbook/recipes/0028.html См. для детализированного объяснения
того, как Rebol открывается, соединяется, и отправляет данные через порты TCP/IP.
Для информации о передаче двоичных файлов и данных непосредственно через

http://compkarori.com/vanilla/display/VID+list+style
http://compkarori.com/vanilla/display/list+examples
http://www.rebol.net/cookbook/recipes/0012.html
http://www.dobeash.com/it/rebdb/
http://softinnov.org/rebol/mysql.shtml


одноранговое подключение, см. http://www.rebol.net/cookbook/recipes/0058.html.

Основная информация о том, как конфигурировать маршрутизаторы и сетевые порты для
использования в этом примере, см. http://portforward.com/. Этот тип конфигурации вне
возможностей этого обучающего материала, но требуется для каждого вида однорангового
("p2p") сетевого приложения. Совместное использование файлов, многопользовательские
сетевые игры, совместное использование веб-камеры, мгновенная передача сообщений,
обслуживание сети, и т.д. Для еще более фундаментального объяснения о том, как сети
работают, и на информацию о том, как конфигурировать типичную сетевую установку в MS
Windows, см. http://com-pute.com/FreeTutorials/Other/NetworkBasics.html.

REBOL [Title: "Одноранговый сетевой менеджер"]

connected: false
; Переменная используется, чтобы узнать действительно ли
; две машины уже соединились.

insert-event-func closedown: func [face event] [
either event/type = 'close [

if connected [
            insert port trim {
                *************************************************

ПОВТОРИТЕ СОЕДИНЕНИЕ.
                ПЕРЕЗАПУСТИТЕ ПРИЛОЖЕНИЕ,
                УДАЛЕННЫЙ ПОЛЬЗОВАТЕЛЬ ОТСОЕДЕНИЛСЯ.
                *************************************************
            }

close port
            if mode/text = "Режим сервера" [close listen]
        ]

quit
    ] [event]
]

view/new center-face gui: layout [
    across

at 5x2 ; этот код устанавливает следующие элементы в GUI

; Текст ниже появляется как опция меню в верхнем левом углу GUI.
; Когда это нажато, текст, содержавшийся в области "display", сохраняется

    ; как файл.

text bold "Сохранить чат" [
filename: to-file request-file/title/file/save trim {

Сохранить как:} "Сохранить" %/c/chat.txt
write filename display/text

    ]

    ; Текст ниже - другая опция меню. Отображает IP адрес пользователя.
; Обращается к серверу, чтобы найти внешний адрес (whatismyip.com).
; Команда "parse" извлекает IP адрес из страницы.

    text bold "Поиск IP" [
        parse read http://whatismyip.com [

thru <title> copy my-ip to </title>]
parse my-ip [

thru "-" copy stripped-ip to end]
        alert to-string rejoin [
            "Внешний: " trim/all stripped-ip "  "
            "Внутренний: " read join dns:// read dns://
        ]
    ]
    ; Текст ниже - третья опция меню. Отображает текст справки.

text bold "Помощь" [
alert {

                Введите IP адрес и число порта в полях.
                Чтобы использовать свой компьютер как сервер,
                нажмите кнопку "Режим сервера".



                Вы должны иметь выставленный IP адрес или
                открытый порт чтобы принимать входящие.
                Выберите "Режим клиента" если вы
                соединяетесь с чьим либо сервером.
                Щелкните "Соеденить", чтобы войти в чат.
                Чтобы проверить приложение на одной машине,
                откройте два приложения, оставьте набор IP
                в "localhost" для обоих. Заставьте одно
                приложение выполняться как
                сервер, а другое как клиент, затем попробуйте соединиться.
        }
    ]
    return

    ; Используемые ниже графические элементы, служат для ввода информации
    ; подключения.

lab1: h3 "IP Адресс:"  IP: field "localhost" 102
lab2: h3 "Порт:" portspec: field "9083" 50
mode: rotary 120 "Режим клиента" "Режим сервера" [

either value = "Режим клиента" [
show lab1 show IP

        ][
hide lab1 hide IP

        ]
    ]

    cnnct: button red "Соеденить" [
hide cnnct
either mode/text = "Режим клиента" [

if error? try [
port: open/direct/lines/no-wait to-url rejoin [

                    "tcp://" IP/text ":" portspec/text]
            ][alert "Сервер не отвечает." return]
        ][

if error? try [
                listen: open/direct/lines/no-wait to-url rejoin [
                    "tcp://:" portspec/text]

wait listen
port: first listen

            ][alert "Сервер активен." return]
        ]

focus entry
        connected: true

forever [
wait port
foreach msg any [copy port []] [

display/text: rejoin [
                    ">>>  "msg newline display/text]
            ]

show display
        ]
    ]

    return  display: area "" 537x500
    return  entry: field 428  ; числа – размеры пикселей

    button "Отправить" [
if connected [

insert port entry/text focus entry
display/text: rejoin [

                "<<<  " entry/text newline display/text]
show display

        ]
    ]
]



show gui do-events ; требуется потому что "/new" используется выше

Чтобы использовать этот пример, пользователь должен знать IP адрес сервера.

17.9 Пример r3D

Этот последний пример - простая демонстрационная версия того, как использовать r3D
модуль Andrew Hoadley's, чтобы использовать трехмерную графику в ваших приложениях.
Для более детализированных и интересных примеров кода, см. примеры в
http://www.rebol.net/demos/download.html.

REBOL [
Title: "Пример r3D"
Notes: {Упрощенный вариант:

        http://www.rebol.net/demos/BF02D682713522AA/i-rebot.r }
]

; Этот раздел загружает сжатую r3D библиотеку:

do to-string decompress 64#{ICltINLdgN+r67fW45+w7fjbG/A38d+Ev0k/+7HstkpnZUGQO6zRbJIL94UsYp4l}

; Этот раздел устанавливает некоторые значения по умолчанию:

base-rot: 150.0 ; степень основного значения вращения по умолчанию
; трансляция камеры и просмотр значений
cameraTransx: 300.0  cameraTransy: 300.0  cameraTransz: 300.0
cameraLookatx: 0.0  cameraLookaty: 0.0  cameraLookatz: 100.0

update: does [
world: copy [] ; обновляет “мир”
; рабочее поле
base-modelWorld: r3d-scale 100.0 100.0 50.0



base-object: reduce [ cube-model base-modelWorld navy ]
append world reduce [ base-object ]
; рабочий стенд
stand-modelWorld: r3d-compose-m4 reduce [

r3d-scale 35.0 35.0 150.0
        r3d-rotatez base-rot
        r3d-translate 0.0 0.0 52.0
    ]

stand-object: reduce [ cube-model stand-modelWorld red ]
append world reduce [ stand-object ]

    ; устанавливает камеру и создает для нее преобразования
    camera: r3d-position-object

reduce [ cameratransx cameratransy cameratransz ]
reduce [ cameraLookatx cameralookaty cameralookatz ]

        [ 0.0 0.0 1.0 ]
    ; Получаем матрицу проекта

Projection: r3d-perspective 250.0
    RenderTriangles: render world camera Projection 400x360
]

RenderTriangles: copy []

out: layout [
r3d-viewport: box 400x360 black effect [draw RenderTriangles]

    across
    style lab label 55 right yellow
    style lab2 label 40 right

vh2 "Вращение:" rbslider: slider 60x16 [
base-rot: (value * 300.0) update show r3d-viewport ]

return
lab "Позиция"
lab2 "x" cpos_x: slider 60x16 [cameratransx:

        (value * 600 - 300.0) update show r3d-viewport]
lab2 "y" cpos_y: slider 60x16 [cameratransy:

        (value * 600 - 300.0) update show r3d-viewport]
lab2 "z" cpos_z: slider 60x16 [cameratransz:

        (value * 600) update show r3d-viewport]
return
lab "Смотреть"
lab2 "x" clook_x: slider 60x16 [cameraLookatx:

        (value * 400 - 200.0) update show r3d-viewport]
lab2 "y" clook_y: slider 60x16 [cameraLookaty:

        (value * 400 - 200.0) update show r3d-viewport]
lab2 "z" clook_z: slider 60x16 [cameraLookatz:

        (value * 200 ) update show r3d-viewport]
]

update
view out

18. Меню

Вот простой прототип, который может быть включен в ваши программы, чтобы обеспечить
дополнительные функциональные возможности меню:

Rebol [Title: "Пример маленького меню"]

view center-face gui: layout/size [

at 100x100 H3 "Вы выбрали:"
display: field

origin 2x2 space 5x5 across
at -200x-200 file-menu: text-list "элемент1" "элемент2" "выход" [

switch value [
            "элемент1" [

face/offset: -200x-200



show file-menu
                ; ПОМЕСТИТЕ СВОЙ КОД ЗДЕСЬ:
                set-face display "Файл / элемент1"
            ]
            "элемент2" [
                face/offset: -200x-200
                show file-menu
                ; ПОМЕСТИТЕ СВОЙ КОД ЗДЕСЬ:
                set-face display "Файл / элемент2"
            ]
            "Выход" [quit]
        ]
    ]

    at 2x2
    text bold "Файл" [

either (face/offset + 0x22) = file-menu/offset [
file-menu/offset: -200x-200
show file-menu

        ][
file-menu/offset: (face/offset + 0x22)

            show file-menu
        ]
    ]

text bold "Помощь" [
file-menu/offset: -200x-200
show file-menu

        ; ПОМЕСТИТЕ СВОЙ КОД ЗДЕСЬ:
set-face display "Помощь"

    ]
] 400x300

Здесь версия файла с расширенными возможностями:
http://www.rebol.org/library/scripts/menu-system.r. Вот минимальный пример:

do http://www.rebol.org/library/scripts/menu-system.r
menu-data: [edit: item "Меню" menu [item "Элемент1" item "Элемент2"]]
reb-style: [item style action [alert item/body/text]]

view center-face layout/size [
at 2x2 menu-bar menu menu-data menu-style reb-style

] 400x500

Демонстрационная версия http://www.rebol.org/library/scripts/menu-system-demo.r
отличается продвинутыми особенностями модуля. Модуль меню был сжат и внедрен в код:

REBOL [Title: "Пример меню"]

; Следующая строка импортирует сжатый модуль меню.

do decompress   #{
789CED7DED761B3792E8EFEBA740343F24AD4D5352EC24C3331E1F59A213258E
E548B233191EDE735A64536A9B6433ECA64566BD8F71DFF7A2AAF0D9F8E82645
27D9DD602672B31B28140A8542A150285C745F9CBF62BD078CA78BECE6B62C3A
8CFD27FE847492CF567378CDF606FBECE8E0E0293BB99D6745992553D69D16E9
58653D1E8F19662DD83C2DD2F9C774F8F881FA7A910E79A97976BD28B37CCA92
E9902D8A94655356E48BF920C537D7D93499AFD8289F4F8A47EC2E2B6F593EC7
7FF345C914AC493ECC46D92001488F58324FD92C9D4FB2B24C876C36CF3F6643
FE50DE2625FF937268E3717E974D6FD8209F0E3328546858507A92961D8DE9FF
AB225BB07C24B11CE4439E7F5194BC8D65C2B1871A92EBFC237C12A45290789A
E66536481FF16C59C1C61C28C032F08056DB48F29A07E3249BA4F3C7319478D5
06B5244ABCF5C30547D38315DB165A8C5A6D821BE683C5249D9689ECDB36EFB6
9C679AB34952A6F32C1917BA67B05F01BAD922ABB1AFD30C0B43A6693249013B



78D68DB9CDC7439E619AEB4C9C266561B5923788E0E7F38223B262D729301D6F9BFBAF2FFF4A0D53DFB163A0C0DD661552566F0B1E28ED7A31EE482130B509F3317E65D24B496CF062DC120A273E17BEC6655ECA48A860EF5E4CD221BB5E395B9B0B50B2FB906052F0CB6A24284F8A4DA79FC738FF97C4FF4BE263FA5398E3717D6C68535054D17ECCFC5530ABB8D170972BFC830FAD326FE1DCE06417AEFFB54621478BAFE94BA3E94BBB5B44BD2D0F13DB048903B65BD132DB18A06BBF4299}

; Ниже пример, как форматировать блок меню.
; Отметьте, что есть два меню в одном блоке.
; Меню "файл" в несколько строк.
; Меню редактирования в одной строке.
; Обратите внимание, что вы можете поместить блоки действия
; после каждого пункта меню:

menu-data: [
file: item "Файл"

menu [
new: item "Элемент 1" [print "Вы выбрали элемент 1"]
open: item "Элемент 2" ; icons [1.png 2.png]

            ---
recent: item "Посмотрите здесь..."

menu [
item "ВЫИГРАЙТЕ ПРИЗ!"
item "Попробуйте дверь, номер два"

                ]



            ---
 exit:   item <Ctrl-Q> "Выход"

          ]
edit: item "Правка" menu [item "копировать" item "вставить"]

]

; Обратите особое внимание на блок действия в меню
; определение стиля ниже. Измените его, чтобы выполнить ваши
; собственные выбранные функции для каждого возможного выбора меню.
; Большая часть определения стиля является полностью дополнительной. Это
; зделанно так, чтобы быть похожим на родное меню Microsoft.
; пример http://www.rebol.org/library/scripts/menu-system-demo.r
; содержит еще много примеров стилей меню и опций.
; Единственная часть требуемая в примере ниже, блок действия "item style":

winxp-menu: layout-menu/style copy menu-data xp-style: [
menu style edge [size: 1x1 color: 178.180.191 effect: none]

color white
        spacing 2x2
        effect none
    item style

font [name: "Tahoma" size: 11 colors: reduce [
black black silver silver]]

colors [none 187.183.199]
effects none
edge [size: 1x1 colors: reduce [none 178.180.191]

effects: []]
action [

            switch/default item/body/text [
                "Выход" [quit]
                "ВЫИГРАЙТЕ ПРИЗ!" [alert "Вы победили!"]
                "Попробуйте дверь, номер два" [alert "Плохой выбор:("]
            ] [print item/body/text]
        ]
]

; Вот простая функция, чтобы задействовать GUI.
; Должна быть включена всякий раз, когда модуль меню используется.

evt-close: func [face event] [
either event/type = 'close [quit] [event]

]
insert-event-func :evt-close

; Финальный пользовательский интерфейс:

window: layout/size [

    ; код ниже показывает меню стиля winxp:

at 2x2 app-menu: menu-bar menu menu-data menu-style xp-style

at 150x200 btn "Кнопка меню" [
        show-menu/offset window winxp-menu
        0x1 * face/size + face/offset - 1x0
    ]
] 400x500

view center-face window

19. Программирование CGI

1) Оболочка: данные могут быть, введены и рассматренны непосредственно в
интерпретаторе Rebol. В коротких сценариях могут использоваться слова "ask" и
"print", чтобы ввести и отобразить данные. В формате "command line"  можно



управлять, вводом текста используя, условные операторы.

Rebol []

forever [
prin "^(1B)[J"
print "Выберите опцию:^/"
print "1 – Время сообщения 1"

    print "2 - Время сообщения 2"
print "3 - Время сообщения 3"
print "4 - Выход^/"

    answer: to-integer ask "Вы выбрали?"
if answer = 4 [ask "^/До свидания!  Нажмите любую клавишу, чтобы закончить."

quit]
print ""
loop answer [print "Rebol лучший!"]
print ""
ask "Нажмите [ENTER] чтобы продолжить"

]

2) GUI: в пределах графического интерфейса, текст введен и отображен в текстовых
полях, областях, и списках. Процессом выполнения программы управляют, используя мыш,
кнопки, списки меню, и другие графические фрагменты:

Rebol []

view layout [
text "Сколько времен должно отображаться сообщение?"
choice "1" "2" "3" [

loop to-integer value [alert "Rebol лучший!"]
    ]
]

3) CGI интерфейс позволяет вводить данные через web-страницу, используя текстовые
поля, области, и всплывающие поля в форме html. Данные поступают в форму, читаются и
обрабатываются в соответствии с программой, которую вы создаете на вашем сервере.
Возвращенные данные отображены как форматированный текст, таблицы, и другие элементы
html, которые динамически созданы и выведены в соответствии с вашей программой.

Два варианта:

1) Html страница:

<HTML><HEAD><TITLE>Форма</TITLE></HEAD><BODY>
<FORM ACTION="http://yourwebserver/yourrebolscript.cgi">
Сколько времен должно отображаться сообщение? <BR>
<INPUT TYPE="TEXT" NAME="times" SIZE="25">
<INPUT TYPE="SUBMIT" NAME="Submit" VALUE="Пуск">
</FORM>
</BODY></HTML>

2) CGI скрипт:

#!/home/youruserpath/rebol -cs
REBOL []
print "content-type: text/html^/"
print [<HTML><HEAD><TITLE>"Форма"</TITLE></HEAD><BODY>]
times: decode-cgi system/options/cgi/query-string
loop to-integer times/2 [print ["Rebol лучший!" <BR>]]
print [</BODY></HTML>]

Все три метода выше могут использоваться, чтобы управлять данными, в онлайн. Вы



можете, например, написать программу, которая использует оболочку или GUI, чтобы
обратиться и управлять данными на сервере. Многопользовательские бизнес приложения,
которые совместно используют базы данных, хорошо подходят для этого типа клиент
серверного дизайна. Каждый пользователь получает копию программы, и все соединяются
с централизованным архивом данных, который существует в одном местоположении в сети.

Чтобы представить данные в Интернете,  вы можете загрузить интерпретатор Rebol на
сервер и написать, cgi приложения в языковом синтаксисе Rebol (выбирают версию Rebol
для операционной системы, выполняющейся на сервере). С CGI пользователи могут
взаимодействовать полностью через знакомый интерфейс web-страницы. Это - один из
самых общих типов компьютерного приложения в современном использовании. И Rebol
облегчает создание этого типа программ.

Требуется знание html,  если вы намереваетесь создавать интернет - приложения. Html
- язык разметки, используемый, чтобы форматировать текст и GUI на web-страницах.
Html не язык программирования, он не позволяет вам управлять данными.

В html элементы на web-странице включены между начальными и конечными тэгами:

<STARTING TAG> Некоторый элемент, который будет включен в web-страницу</ENDING TAG>

Плужирный текст:

<STRONG> полужирный текст</STRONG>

Чтобы создать таблицу с тремя строками данных, сделайте следующее:

<TABLE border=1>
<TR><TD>Первая строка</TD></TR>
<TR><TD>Вторая строка</TD></TR>
<TR><TD>Третья строка</TD></TR>
</TABLE>

Если хотим отобразить текущее время и дату в теле web-страницы:

#!/home/youruserpath/rebol -cs
REBOL []
print "content-type: text/html^/"
print [<HTML><HEAD><TITLE>"Тиры"</TITLE></HEAD><BODY>]
print ["Текущая дата и время: " now]
print [</BODY></HTML>]

Код выше содержит встроенный слово "now", которое выводит время и дату на веб
странице.

Пример ниже - короткая программа, которая читает и сортирует информационные
наполнения текстового файла, названного "users.txt", и отображает его в вашей web-
странице:

#!/home/youruserpath/rebol -cs
REBOL []
print "content-type: text/html^/"
print [<HTML><HEAD><TITLE>"Титры"</TITLE></HEAD><BODY>]
if exists? %users.txt [data: sort read/lines %users.txt print data]
print [</BODY></HTML>]



Если вы хотите, чтобы пользовательские данные были вводимы в вашу программу CGI, вы
должны создать страницу html с "form". Формы Html включают текстовые поля ввода,
всплывающие рамки выбора и другие графические фрагменты, которые учитывают ввод
данных. Форма должна содержать единственное "action", которое связывается с адресом
сети вашей CGI программы. Шаблон формы ниже содержит текстовое поле ввода и
действие, которое указывает на http://yourwebserver/yourrebolscript.cgi.

<HTML><HEAD><TITLE>Моя форма</TITLE></HEAD><BODY>
<FORM ACTION="http://yourwebserver/yourrebolscript.cgi">
<INPUT TYPE="TEXT" NAME="username" SIZE="25">
<INPUT TYPE="SUBMIT" NAME="Submit" VALUE="Пуск">
</FORM>
</BODY></HTML>

Чтобы извлечь информацию, включите следующий код в свою программу CGI:

a-word: decode-cgi system/options/cgi/query-string

Вы можете использовать назначенное выше слово, чтобы обратиться ко всем посланным
данным. Rebol автоматически назначает слова на ввод данных в форме html. Если вы
напечатаете "Fred Thompson" как название в текстовом поле входа в форме выше, то
Rebol декодирует это как:

[username: "Fred Thompson"]

Вы можете использовать эти данные в программе CGI следующим образом:

#!/home/youruserpath/rebol -cs
REBOL []
print "content-type: text/html^/"
print [<HTML><HEAD><TITLE>"Титры"</TITLE></HEAD><BODY>]
username: decode-cgi system/options/cgi/query-string
print ["Привет " second username "!"]
print [</BODY></HTML>]

В коментариях не нуждается.

Программа выводит простую страницу, которая отображает имена всех ее прежних
посетителей, позволяет вам вводить свое имя, и затем называет себя, когда вы
нажимаете, запускает процесс снова:

#!/home/youruserpath/rebol -cs
REBOL []
print "content-type: text/html^/"
print [<HTML><HEAD><TITLE>"Моя страница"</TITLE></HEAD><BODY>]
print ["Вот список посетителей, которые побывали здесь:" ]
print [<BR>] ; prints a carriage return
either exists? %users.txt [
    newname: decode-cgi system/options/cgi/query-string
    if newname/2 <> none [
        write/append %users.txt join "<BR>" [newname/2]

]
    signatures: read %users.txt
    print signatures
] [write %users.txt ""]
print [<FORM ACTION="http://yourwebserver/yourrebolscript.cgi">]
print [<BR><HR><BR>"Введите имя:"<BR>]



print ["Имя: "<INPUT TYPE="TEXT" NAME="username" SIZE="25">]
print [<INPUT TYPE="SUBMIT" NAME="Submit" VALUE="Пуск">]
print [</FORM>]
print [</BODY></HTML>]

Если вы помните синтаксис предыдущих двух примеров, вы находитесь на верном пути к
тому, чтобы быть способным создавать динамические приложения для Интернета с Rebol.
Шаблоны с примерами, являются базой для создания CGI приложений независимо от того,
какой комплекс программ вы создаете:

1. Проектируйте web-страницу с формой html, которая получает ввод от
пользователя. Укажите действие формы к url вашего сценария CGI.

2. Проектируйте CGI  программу, чтобы обработать ввод данных через форму. Первые
три строки должны быть такие же как в шаблоне. Используйте "decode-cgi
system/options/cgi/query-string", чтобы автоматически назначить переменные
слова на данные.

Информация по Rebol CGI программированию:

http://rebol.com/docs/cgi1.html

http://rebol.com/docs/cgi2.html

http://rebol.com/docs/cgi-bbs.html

http://www.rebol.net/cookbook/recipes/0045.html

Сайт ниже содержит бесплатную  программу сервера, написанную полностью на Rebol:

http://plain.at/vpavlu/plain-dev/r80v5/r80v5.html

Это обеспечивает альтернативу программирования CGI, который позволяет вам включать
код Rebol непосредственно в ваши страницы html, подобно языку программирования PHP.
Используя вышеупомянутый сервер, вы можете включить код Rebol, который будет
работать непосредственно в ваших web-страницах, используя следующий синтаксис:

<?rebol {Код rebol} ?>

Вот пример:

<HTML>
<BODY>
Текущее время: <?rebol print now/time ?>.
</BODY>
</HTML>

CGI интерфейс даже не требуется. Это - удобная опция.

20. Parsing или синтаксический анализ

Встроенная функция "parse" - важная часть языка Rebol. Используется, чтобы
импортировать и преобразовать организованные участки памяти внешних данных в блочный
формат, который понимает Rebol. Также обеспечивает средство разделения, поиска,
сравнения, распаковки, и распределения в организованную информацию из бесформатных
текстовых данных. Его способности сопоставления с образцом, подобны регулярным
выражениям в языке Perl и других языках.

Основной формат для синтаксического анализа:

http://rebol.com/docs/cgi1.html
http://rebol.com/docs/cgi2.html
http://rebol.com/docs/cgi-bbs.html
http://www.rebol.net/cookbook/recipes/0045.html
http://plain.at/vpavlu/plain-dev/r80v5/r80v5.html


parse <data> <matching rules>

У синтаксического анализа есть несколько режимов использования. Самый простой режим,
разбивает текст разделителями и преобразовывает эти части в блок Rebol. Чтобы
сделать это, не определяйте "none" как правило. Общие разделители, запятые, вкладки,
точки с запятой, и newlines. Вот некоторые примеры:

text1: "яблоко апельсин персик"
parsed-block1: parse text1 none

text2: "яблоко,апельсин,персик"
parsed-block2: parse text2 none

text3: "яблоко       апельсин               персик"
parsed-block3: parse text3 none

text4: "яблоко;апельсин;персик"
parsed-block4: parse text4 none

text5: "яблоко,апельсин персик"
parsed-block5: parse text5 none

text6: {"яблоко","апельсин","персик"}
parsed-block6: parse text6 none

text7: {
яблоко
апельсин
персик
}
parsed-block7: parse text7 none

Все вышеупомянутые анализируемые блоки оценивают как ["яблоко" "апельсин" "персик"].
Это полезно, потому что возможно простое импортирование файлов данных, созданных
другими программами. Вы могли, например, использовать правило "none" чтобы
импортировать файлы CSV, созданные приложением базы данных или электронной таблицей.
Данные автоматически преобразованы к блокам данных Rebol для использования в ваших
сценариях.

Если надо разбить файлы, основанные на некотором символе кроме общих разделителей,
вы можете определить разделитель, как правило:

text: "яблоко*апельсин*персик"
parsed-block: parse text "*"

text: "яблоко&апельсин&персик"
parsed-block: parse text "&"

text: "яблоко    &    апельсин&персик"
parsed-block: parse text "&"

Вы можете также включить смешанные множественные символы, которые будут
использоваться как разделители:

text: "яблоко&апельсин*персик"
parsed-block: parse text "&*"

text: "яблоко&апельсин*персик"
parsed-block: parse text "*&" ; порядок не имеет значения



Используя режим "splitting" синтаксического анализа отличный способ получить
отформатированные таблицы данных.

Разбивая текст переводами каретки, вы сталкиваетесь с небольшой проблемой:

text: { Фамилия
        Имя
        Улица
        Город, Страна }

parsed-block: parse text "^/"

; ^/ символ в Rebol для перевода каретки

Места включены в правило синтаксического анализа по умолчанию, таким образом, вы
получаете совокупность данных, это более логично, чем ранее:

["Фамилия" "Имя" "Улица" "Город" "Страна"]

Вы можете использовать "/all" , чтобы устранить места из правила разделителя. Код
ниже:

text: { Фамилия
        Имя
        Улица
        Город, Страна }

parsed-block: parse/all text "^/"

преобразовывает данный текст в следующий блок:

[" Фамилия" " Имя" " Улица"
    " Город, Страна "]

Теперь вы можете урезать дополнительное пространство в каждой строке:

foreach item parsed-block [trim item]

и получим это:

["Фамилия" "Имя" "Улица" "Город, Страна"]

Пример выше мог быть откорректирован, чтобы переместить восстановленные строки
информации из любого текстового файла в ваше приложение. Но это не может сделать
весь синтаксический анализ. Вы можете также использовать это, чтобы проверить,
существуют ли какие-нибудь определенные данные в пределах данного блока. Чтобы
сделать это, определите правило как элемент, который вы ищете. Вот пример:

parse ["яблоко"] ["яблоко"]



parse ["яблоко" "апельсин"] ["яблоко" "апельсин"]

ВАЖНО: По умолчанию, как только синтаксический анализ сталкивается с тем, что не
соответствует, все выражение оценивает как лож, ДАЖЕ ЕСЛИ данное правило найдено
однин или более раз в данных. Например, следующее - ложь:

parse ["яблоко" "апельсин"] ["яблоко"]

Такое поведение только значения по умолчанию. Вы можете управлять, как
синтаксический анализ должен отвечать на элементы, которые не соответствуют.
Добавление слов ниже к правилу возвратит истину, если данное правило будет
соответствовать данным:

1. "any" - правило соответствует нулю данных или больше раз
2. "some" - правило соответствует данным один или более раз
3. "opt" - правило соответствует нулю данных или одно время
4. "one" - правило соответствует данным точно одно время
5. an integer - правило соответствует данным данное число раз
6. two integers - правило соответствует данным, неоднократно включенным в

диапазон между этими двумя целыми числами

Следующие примеры - истина:

parse ["яблоко" "апельсин"] [any string!]
parse ["яблоко" "апельсин"] [some string!]
parse ["яблоко" "апельсин"] [1 2 string!]

Вы можете создать правила, которые включают множественные опции соответствия -
только отделяют выборы символом "|"  и заключают их в скобках. Следующее истинно:

parse ["яблоко" "апельсин"] [any [string! | url! | number!]]

Вы можете включить действия, чтобы они срабатывали всякий раз, когда соответствуют
правилу. Заключите действие в круглые скобки:

parse ["яблоко" "апельсин"] [any [string!
    (alert "Блок содержит строку.") | url! | number!]]

Вы можете перескочить через данные, игнорируя участки памяти, пока не добираетесь до
или мимо данного условия. Слово "to" игнорирует данные, ПОКА условие не найдено.
Слово "thru" игнорирует данные, пока ТОЛЬКО МИМО условия. Следующее истинно:

parse [234.1 $50 http://rebol.com "яблоко"] [thru string!]

Реальное значение сопоставления с образцом - то, что вы можете искать и извлечь
данные из восстановленного текста, организованным способом. Слово "copy"
используется, чтобы назначить переменную на соответствующие данные. Например,
следующий код загружает необработанный html, игнорирует все кроме текста между
тэгами:

parse read http://rebol.com [
thru <title> copy parsed-text to </title> (alert parsed-text)



]

Следующий код расширяет пример выше, чтобы обеспечить полезную особенность показа
внешнего IP адреса компьютера. Читает http:/whatismyip.com, анализирует текст
заголовка, и затем анализирует тот текст снова, чтобы возвратить только номер IP.
Сетевой адрес также отображен, используя встроенный в Rebol dns протокол:

parse read http://whatismyip.com [
thru <title> copy my-ip to </title>

]
parse my-ip [

thru "-" copy stripped-ip to end
]
alert to-string rejoin [
    "Внешний: " trim/all stripped-ip "  "
    "Внутренний: " read join dns:// read dns://
]

Вот полезный пример, который удаляет все комментарии из данного сценария Rebol
(любая часть строки, которая начинается с точки с запятой ";"). Этот код основан на
сценарии в http://www.rebol.org/library/scripts/uncomment.r. Сначала просит имя
файла, и назначает информационные наполнения того файла к переменному слову "code".
Используя слова "to" и "thru" , информационное наполнение тогда анализируется для
любого текста, который начинается с символа ";", и до конца строки. Блок действия
использует правила синтаксического анализа, встроенную функцию "remove/part", чтобы
удалить не нужное. Наконец, анализируемый код, посылают встроенному редактору текста
Rebol для того, чтобы рассмотреть, сохранить, и т.д.:

code: read to-file request-file
parse/all code [any [

to #";" begin: thru newline ending: (
remove/part begin ((index? ending) - (index? begin))) :begin

    ]
]
editor code

Чтобы узнать больше, см. следующие ссылки:

http://www.codeconscious.com/rebol/parse-tutorial.html

http://www.rebol.com/docs/core23/rebolcore-15.html

http://en.wikibooks.org/wiki/REBOL_Programming/Language_Features/Parse

http://www.rebolforces.com/zine/rzine-1-06.html#sect4.

21. Объекты Rebol

Объекты Rebol - структуры кода, которые позволяют вам формировать и копировать
пакеты функций и данных вместе так, чтобы они могли быть легко скопированы,
многократно использованы, и сохранены как модули. Объекты Rebol обеспечивают простой
способ проектировать шаблоны многократного использования связанного кода, который
может быть повторен и изменен, чтобы создать новые модули кода. Объекты также
обеспечивают способ создать аналоги объектов, так, чтобы любой объектный код мог
служить основанием для нового объектного дизайна. Таким образом, понятие объектной
имитации может быть расширено для множественных поколений, чтобы создать изменения
кода, полученного из единственного шаблона.

Объекты, прежде всего, полезны в Rebol, для создания дубликатов структур данных и
копий кода. Объекты также полезны тем, что они скрывают внутренние данные и функции
в пределах их структуры. Они могут гарантировать, что слова, используемые в одном

http://www.codeconscious.com/rebol/parse-tutorial.html
http://www.rebol.com/docs/core23/rebolcore-15.html
http://en.wikibooks.org/wiki/REBOL_Programming/Language_Features/Parse
http://www.rebolforces.com/zine/rzine-1-06.html#sect4.


месте кода, по ошибке не используются в другом месте в пределах той же самой
программы. Руководящее использование слова таким образом (скрывающиеся и
выставляющие определения слова в пределах различных разделов программы) упоминается
как управление "namespace".

Отметьте: В отличие от других языков программирования, в которых большинство кода
задумано и организовано в терминах объектных конструкций, Rebol обеспечивает
маленькое подмножество инструментальных средств, которые позволяют вам просто
имитировать проекты кода. Классы - фундаментальная постройка, в других объектно-
ориентированных языках, но они толичаются в Rebol. Если вы знакомы с другими
объектно-ориентированными языками программирования, то знайте что объектная модель
("class") свойства в Rebol просто скопирована, а не наследована. Различие - то, что
изменения в "родительской - parent" объектной модели прототипа в Rebol не передают
ее полученным дочерним объектам, когда они созданы.

Чтобы сделать объектный проект в Rebol, используйте следующий синтаксис:

label: make object! [object definition]

Объектное определение может содержать функции, значения, или данные любого типа.

Вот пример, который определяет пустой объект учетной записи пользователя. Содержит 6
переменных слов (отметьте, что каждое слово расположено каскадом вниз, чтобы не
равняться "none"):

account: make object! [
first-name:
last-name:
address:

    phone:
email-address:
none

]

Определение учетной записи выше просто упаковывает эти 6 переменных в контейнер, или
контекст, названный "account".

Вы можете обратиться к данным и функциям в пределах объекта, используя
усовершенствование ("/ path"):

object/word

В примере выше, "account/phone" обращается к данным телефонного номера,
содержавшимся в учетной записи. Вы можете сделать изменения элементов в объекте
следующим образом:

object/word: data

В примере выше:

account/phone: "555-1234"
account/address: "4321 Москва Останкино, Россия 54321"

На многих обьектных языках примечание "dot" используется, чтобы обратиться к
элементам, содержавшимся в объекте (то есть, "methods" и "properties"), используя



формат "object.method".

Как только объект создан, вы можете рассмотреть все его информационные наполнения,
используя встроенное слово "help":

help object

; or

? object

; "?" синоним для слова "help"

Если пример набрали в интерпритаторе Rebol, то увидите только это:

? account

Отобразит информацию:

ACCOUNT is an object of value:
   first-name      none! none
   last-name       none! none
   address         string!   "4321 Москва Останкино, Россия 54321"
   phone           string!   "555-1234"
   email-address   none! none

Как только вы создали объектный прототип, вы можете сделать новый объект основанным
на оригинальном определении прототипа. Новый объект будет содержать все данные и
функции, определенные в оригинальном прототипе. Чтобы создать такой имитируемый
объект, используйте следующий синтаксис:

label: make existing-object [
значения, которые будут изменены из оригинального определения прототипа

]

Например, код ниже создает метку блока "user1" новой учетной записи. Вы можете
изменить любую часть оригинального объектного блока, переопределить переменные, и
т.д., следующим образом:

user1: make account [
first-name: "Джон"
last-name: "Смит"
address: "4321 Москва Останкино, Россия 54321"
email-address: "john@hisdomain.com"

]

Обратите внимание, что переменная телефонного номера не установлена в вышеупомянутом
примере. В этом случае, переменная телефонного номера сохраняет значение по
умолчанию "none" установленного в оригинальном определении учетной записи. Это один
из ключевых моментов использования объектов.

В дополнение к изменению существующих переменных в объектном определении, вы можете
также расширить существующее объектное определение с новыми значениями:



label: make existing-object [new-values]

Определение ниже создает новый объект учетной записи, переопределяет все
существующие переменные, и прилагает новую переменную.

user2: make account [
first-name: "Сергей"
last-name: "Михаил"
address: "4321 Москва Останкино, Россия 54321"
phone:  "555-1234"
email-address: "bob@mysite.net"
favorite-color: "красный"

]

"user2/favorite-color" теперь обращается к "красный" ("user1/favorite-color" все еще
не определен - вы получите ошибку, если вы попытаетесь обратиться к "user2/favorite-
color").

Код ниже создает дубликат учетной записи user2, с только измененным названием:

user2a: make user2 [
first-name: "Павел"
email-address: "paul@mysite.net"

]

"? user2a" покажет следующую информацию:

USER2A is an object of value:
   first-name      string!   "Павел"
   last-name       string!   "Михаил"
   address         string!   "4321 Москва Останкино, Россия 54321"
   phone           string!   "555-1234"
   email-address   string!   "paul@mysite.net"
   favorite-color  string!   "красный"

Вы можете создать любое число параметров пользователя, основанных на шаблоне
"account". Любое поле, которое не заполнено, сохранит значение по умолчанию "none",
установленного в определении шаблона.

С вышеупомянутыми пользовательскими определенными объектами вы можете выполнить
сравнения или любые другие операции, которые ссылаются на включенные данные, такие
как:

if user1/address = user2/address [
    print "Оба пользователя соседи."
]

Вы можете также включить функции в своем объектном определении:

account: make object! [
first-name:

    last-name:
    address:
    phone:
    none



    email-address: does [
return to-email rejoin [

first-name "_" last-name "@website.com"
        ]
    ]
    display: does [

 print ""
print rejoin ["Имя:     " first-name " " last-name]
print rejoin ["Адрес:  " address]
print rejoin ["Телефон:    " phone]
print rejoin ["Email:    " email-address]
print ""

    ]
]

user1: make account []

user2: make account [
first-name: "Сергей"
last-name: "Михаил"
phone:  "555-4321"

]

user3: make account [
first-name: "Боб"
last-name: "Джонс"
address: "4321 Москва Останкино, Россия 54321"
phone:  "555-1234"
email-address: "bob@mysite.net"

]

; распечатать все данные, содержавшиеся в каждом объекте:

user1/display user2/display user3/display

Важно видеть здесь, что оригинальное определение объекта учетной записи включает
функцию "display". Прототип учетной записи также определяет все переменные, чтобы не
быть "none" по умолчанию. Каждый новый объектный экземпляр класса (user1, user2,
user3) содержит весь тот код по умолчанию. В результате объект user1, например,
является полностью функциональным объектом даже при-том, что никакие переменные или
функции не были определены во время его создания. Все, что требуется, чтобы создать
объект, использовать "user1: make account []".

Также важно распознать, что в каждом пользовательском экземпляре класса, переменные,
показанные функцией display, являются временными данными к приведенному примеру.
Например, когда вызывают функцию user3/display, она печатает переменные,
содержавшиеся в объекте user3. Ни одна из переменных, содержавшихся в user1 или
user2, не перепутанна, даже при-том, что они упомянуты тем же самым словом. Это одно
из приемуществ объектов, группа переменных и функций формируется в легко повторяемые
объекты со своим содержанием. Переменные, содержавшиеся в любом из объектов, могут
быть обновлены в любое время. Отметьте, что переменная "email-address" первоначально
назначена на функцию, которая определяет адрес электронной почты значения по
умолчанию, "first-name_last-name@website.com". Вы можете отменить это определение,
как пользователь, просто обратиться к строке адреса электронной почты. Как только вы
сделали это, функция адреса электронной почты больше не существует.

Следующий пример показывает пользу обьектов. Это маленькая игра, в которой
множественные символьные объекты созданы из дублированного шаблона. Каждый объект
содержит некоторый код и данные, которые позволяют пользователю перемещать несколько
воображаемых символов. Основанная на пользовательском выборе, текущая позиция
символа обновляется. Созданы пять экземпляров класса символов, каждый с различным
начальным свойством позиции. Игрок должен найти скрытый приз и заставить один из
символов двигаться в его местоположение. После каждого пользовательского перемещения
выяснена позиция символьного объекта, если это соответствует позиции скрытого приза:

Rebol []
; Создайте случайную координату в который



; будет найден скрытый приз:

hidden-prize: random 15x15

; Вот основное "character" определение прототипа.
; Это содержит переменную, которая держит символ
; текущей позиции. Это также содержит функцию которая
; получает выбор движения от пользователя, затем обновляет
; текущую позицию символа, и сообщает,
; если найден скрытый приз .

character: make object! [
position: 0x0

    move: does [
direction: ask "Верх, Низ, Лево, Право:  "

        ; меняет позицию объекта, основанную на выборе пользователя
switch/default direction [

            "верх" [position: position + -1x0]
            "низ" [position: position + 1x0]
            "лево" [position: position + 0x-1]
            "право" [position: position + 0x1]
        ] [print newline print "не верное указание!"]
        ; отобразит сообщение, если найден скрытый приз:

if position = hidden-prize [
print newline
print " Вы нашли скрытый приз. ВЫ ПОБЕДИТЕЛЬ!"
print newline
halt

        ]
print rejoin [

newline
            " Вы переместили символ " movement " " direction

".  Символ" movement "теперь"
            hidden-prize - position
            " далеко от скрытого приза. "

 newline
        ]
    ]
]

; Теперь созданы объекты на пять символов, каждый с
; различной начальной позицией:

character1: make character[]
character2: make character[position: 3x3]
character3: make character[position: 6x6]
character4: make character[position: 9x9]
character5: make character[position: 12x12]
; Пользователь получает 20 возможностей. Во время каждого шанса,
; выполняется функция выбранного символа и
; отображается текущая позиция всех символов.

loop 20 [
prin "^(1B)[J"
movement: ask " Какой символ вы хотите переместить (1-5)? "
if find ["1" "2" "3" "4" "5"] movement [

do rejoin ["character" movement "/move"]
print rejoin [

newline
            " Позиция каждого символа сейчас:  "
            newline newline
            "СИМВОЛ ОДИН:   " character1/position newline
            "СИМВОЛ ДВА:   " character2/position newline
            "СИМВОЛ ТРИ: " character3/position newline
            "СИМВОЛ ЧЕТЫРЕ:  " character4/position newline
            "СИМВОЛ ПЯТЬ:  " character5/position
        ]

ask "^/ Нажмите клавишу [Enter], чтобы продолжить."
    ]



]

Другое использование объектов должно избежать столкновений пространства имен,
которые происходят в результате использования того же самого слова в больше чем
одном контексте. Поскольку вы пишете больше кода, вы можете создать библиотеку
функций для общих задач. Если вы используете то же самое слово в двух функциях,
чтобы представить отдельные действия, они могут вызвать проблему, если вы
импортируете и используете функции в той же самой программе. Если это случается,
одна функция может неумышленно переопределить слово, которое вы используете в другой
функции, и заставляете ее выступать неожиданным способом. Простое решение такого
беспорядка пространства имен состоит в том, чтобы обернуть ваши библиотечные функции
в объекте, и выставить служебные слова, которые вы хотите в своей программе. Таким
образом, вы можете назвать функцию в контексте названия объекта, и не волноваться о
столкновениях пространства имен. Например, в коде ниже, первая функция "bank" и
"var" переменная записаны поверх, когда те слова переопределены во второй раз:

Rebol []
var: 1234.56
bank: does [

print ""
print rejoin ["Ваш счет в банке:  $" var]
print ""

]

var: "КЕДР"
bank: does [

print ""
print rejoin [

        " Ваш любимый банк:  " var]
print ""

]

bank

Оценивая "bank" в конце кода "Ваш любимый банк:  КЕДР". Нет способа обратиться к
коду счета в банке в этом пункте, потому что слова были полностью переопределены.
Если бы такое переопределение слов было сделано не умышленно в каком-нибудь коде,
оно, вероятно, вызвало бы ошибку. Вы можете избежать проблем, просто обертывая
вышеупомянутый код в отдельные объекты:

Rebol []
show-money: make object! [

var: 1234.56
bank: does [

print ""
print rejoin ["Ваш счет в банке:  $" var]
print ""

    ]

]

show-place: make object! [
var: "КЕДР"
bank: does [

        print ""
print rejoin [

            "Ваш любимый банк:  " var]
print ""

    ]
]

show-money/bank
show-place/bank



Функция "bank" может использовать переменную "var" тем способом, которым должным
образом связано с его контекстом. Обертывание кода в объектах обеспечивает простой
метод организации, группирования, и поддержания переменных и функций, которые не
должны быть изменены другими частями программы.

Чтобы просмотреть, как вышеупомянутые объекты могли быть имитированы и расширены,
объекты ниже создают GUI, которые изменяют информационные наполнения переменных и
используют функции, содержавшиеся в вышеупомянутых объектах. Добавте код ниже к выше
показанному:

deposit: make show-money [
view layout [

button "Депозит $" [
var: var + 10
bank

        ]
    ]
]

travel: make show-place [
view layout [

new-favorite: field 300 trim {
Наберите название банка, и нажмите [Enter]} [
var: value
bank

        ]
    ]
]

Как полученный объект, у объекта "deposit" автоматически есть доступ ко всем данным
и функциям, содержавшимся в объекте показа денег, и объект "travel" получает полный
доступ к информационным наполнениям объекта. В каждом новом объекте блок действия
графического фрагмента GUI изменяет переменную "var" и выполняет действие "bank",
наследованное от объектов, созданных ранее.

Поскольку вы исследуете Rebol, вы найдете, что большая часть языка основана на
объектных конструкциях. Работа с графическими символами часто требует обширной
манипуляции с объектами. Название самого языка означает "Relative Expression Based
Object Language". Этим сказанно, объектно-ориентированные понятия не столь важны в
Rebol, как в других языках. Суть Rebol состоит в том, чтобы организовать данные и
функции, используя блоки, и обратиться к блокам со словами. Факт, что любой блок
может автоматически быть обработан как порядковая прогрессия и управляем
использующимися встроенными особенностями манипуляции списка, является
фундаментальным понятием идеи Rebol. Другое отличие, которое существенно отделяет
Rebol от других  объектно-ориентированных языков, то, что языковые диалекты могут
быть основаны на определениях слова, обеспечивая естественный интерфейс данным и
функциям. Объекты просто обеспечивают дополнительную возможность формировать,
копировать код и расширить встроенные объектные конструкции, которые уже существуют
в языке.

Для получения дополнительной информации о том, как использовать объекты Rebol , см.:

http://rebol.com/docs/core23/rebolcore-10.html

http://en.wikibooks.org/wiki/REBOL_Programming/Language_Features/Objects

http://en.wikipedia.org/wiki/Prototype-based_programming

22. Общие ошибки

Ниже показанны решения общих ошибок, с которыми вы столкнетесь, экспериментируя с
Rebol:

1) "** Syntax Error: Script is missing a REBOL header" - Всякий раз, когда вы

http://rebol.com/docs/core23/rebolcore-10.html
http://en.wikibooks.org/wiki/REBOL_Programming/Language_Features/Objects
http://en.wikipedia.org/wiki/Prototype-based_programming


создаете сценарий, он должен содержать минимальный необходимый заголовок наверху
кода. Включайте следующий текст в начале сценария:

Rebol []

2) "** Syntax Error: Missing ] at end-of-script" - Вы получите эту ошибку, если не
поместите закрывающую квадратную скобку в конце блока. Вы будете видеть подобную
ошибку для открытых строк. Код ниже даст вам ошибку, потому что отсутствует "]" в
конце блока:

fruits: ["апельсин" "яблоко" "груша" "виноград"
print fruits

ниже правильно:

fruits: ["апельсин" "яблоко" "груша" "виноград"]
print fruits

Выравнивание блоков помогает находить и устранять эти виды ошибок.

3) "** Script Error: request expected str argument of type: string block object
none" - Этот тип ошибки происходит, когда вы пытаетесь передать неправильный тип
значения к функции. Код ниже даст вам ошибку, потому что Rebol автоматически
интерпретирует переменную вебсайта как url, и функция "alert"  требует значения
строки:

website: http://rebol.com
alert website

Код ниже решает проблему, преобразовывая значение url в строку прежде, чем передать
это к функции "alert":

website: to-string http://rebol.com
alert website

4) "** Script Error: word has no value" - Проверки правописания выявится этот тип
ошибки. Вы можете столкнуться с этим в любое время, если пытаетесь использовать
слово, которое не определено или зарезервированно в интерпретаторе Rebol, или вами,
в предыдущем коде:

wrod: "Hello world"
print word

5) ВАЖНО: Вот причуда Rebol, которая не выявляет ошибку, но может вызвать
запутывающие результаты, особенно если вы знакомы с другими языками:

unexpected: [
empty-variable: ""
append empty-variable "*"
print empty-variable

]



do unexpected
do unexpected
do unexpected

Строка:

empty-variable: ""

повторно не инициализирует переменную к пустому статусу. Вместо этого каждый раз,
когда блок выполнен, "empty-variable" содержит предыдущее значение. Чтобы задержать
переменную, как предназначено, используют слово "copy" следующим образом:

unexpected: [
empty-variable: copy ""
append empty-variable "*"
print empty-variable

]

do unexpected
do unexpected
do unexpected

6) Load/Save, Read/Write, Mold, Reform, etc - другой пункт ошибок, с которым вы
можете столкнуться в Rebol. Они имеют отношение к различным словам, которые читают,
пишут, и форматируют данные. Сохраняя данные в файл на жестком диске, вы можете
использовать "save" или "write". "Save" используется, чтобы хранить данные в
формате, более пригодном для использования в Rebol. "Write" сохраняет данные как
есть. "Load" и "read" сопоставимые отношения. "Load" читает данные пути ввода.
"Read" открывает данные в формате, как есть, байт в байт. Для получения
дополнительной информации, см. следующие словарные статьи Rebol:

http://rebol.com/docs/words/wload.html

http://rebol.com/docs/words/wsave.html

http://rebol.com/docs/words/wread.html

http://rebol.com/docs/words/wwrite.html

Другие встроенные слова, такие как "mold" и "reform" помогают вам работать с текстом
такими способами, которые являются более удобочитаемыми интерпретатором Rebol. Для
полезного объяснения, см. http://www.rebol.net/cookbook/recipes/0015.html.

7) Порядок очередности - выражения Rebol всегда оцениваются слева направо,
независимо от вовлеченных операций. Если вы хотите, чтобы определенные
математические операторы были оценены первыми, они должны или быть включены в
круглую скобку или помещены в начале выражения. Например:

2 + 4 * 6

то же самое как:

(2 + 4) * 6 ; сначала вычисленна  левая часть

== 6 * 6

http://rebol.com/docs/words/wload.html
http://rebol.com/docs/words/wsave.html
http://rebol.com/docs/words/wread.html
http://rebol.com/docs/words/wwrite.html
http://www.rebol.net/cookbook/recipes/0015.html


== 36

Это противоречит другим знакомым правилам оценки. В других языках, например,
умножение обычно обрабатывается перед суммированием. Так, то же самое выражение:

2 + 4 * 6

обработан как:

2 + (4 * 6) ; сначала выполнен оператор умножения
== 2 + 24

== 26

В Rebol порядок старшинства с лева на право является естественным. Признаки конца
строки даже не требуются, как в других языках (точки с запятой в C-образных),
смотрите следующий пример:

save %text.txt read http://rebol.com print sort read %text.txt

может быть написан более интуитивно как это:

save %text.txt (read http://rebol.com)
print (sort (read %text.txt))

Этот сценарий оценен следующим образом: функция "save" принимает два параметра -
сначала имя файла, и данные, которые будут записаны. Те данные читаются из
http://rebol.com. Затем функция "print" берет один параметр. Данные - из text.txt
файла.

22.1 Ловушка ошибок

Есть несколько простых способов препятствовать, программе терпеть крах, когда
происходит ошибка. Слова "error?" и "try" вместе, обеспечивают проверку и обработку
ожидаемых ошибочных ситуаций. Например, если интернет-подключение не будет доступно,
то код ниже потерпит крах:

html: read http://rebol.com

Откорректированный код ниже обработает ошибку более изящно:

if error? try [html: read http://rebol.com] [
alert "Недоступен."

]

Слово "attempt" альтернатива подпрограммы "error? try". Это возвращает оцененные
информационные наполнения данного блока, если это успешно. Иначе возвращает "none":

if not attempt [html: read http://rebol.com] [
alert "Недоступен."



]

Дополнительный материал: http://www.rebol.com/docs/core23/rebolcore-17.html.

23. Проектирование программ

Попытка заняться программированием на пустом месте требует не только понимания того,
как языковые компоненты работают и как они соединены, но и думать в терминах
организации тех компонентов рабочей программы из вашего предполагаемого проекта.
Пока вы не сделали это много раз, может быть трудно, сформировать программу из мысли
и необработанного кода. Этот текст предназначен, чтобы дать немного понимания того,
как вы можете приблизиться к организации ваших мыслей и написать код, чтобы
удовлетворить ваши определенные потребности для любой ситуации.

Отличный способ начать писать код проекта состоит в том, чтобы продумать блок-схему
деятельности и выписать все необходимые требования кодирования в схеме дизайна,
состоящей из естественного языка. Используем разговорный английский псевдокод, для
детализации кода, который вы напишете. Описать в общих чертах, как программа будет
работать, выделить структуру программы, привести к псевдокоду в пределах той
структуры,  далее к конечному детализированному коду. Как последний шаг, отладить
код, добавить или изменить функциональные возможности.

Удачи Вам!

Copyright © Nick Antonaccio 2005-2006, All Rights Reserved

Попытка перевода: Andrix, Andrix7@yandex.ru.

